

# CIMP

Peter Gammie

February 23, 2021

## Abstract

CIMP extends the small imperative language IMP with control non-determinism and constructs for synchronous message passing.

## Contents

<b>1</b>	<b>Point-free notation</b>	<b>1</b>
<b>2</b>	<b>Infinite Sequences</b>	<b>3</b>
2.1	Decomposing safety and liveness . . . . .	6
<b>3</b>	<b>Linear Temporal Logic</b>	<b>9</b>
3.1	Leads-to and leads-to-via . . . . .	15
3.2	Fairness . . . . .	18
3.3	Safety and liveness . . . . .	20
<b>4</b>	<b>CIMP syntax and semantics</b>	<b>22</b>
4.1	Syntax . . . . .	22
4.2	Process semantics . . . . .	23
4.3	System steps . . . . .	24
4.4	Control predicates . . . . .	25
<b>5</b>	<b>State-based invariants</b>	<b>27</b>
5.0.1	Relating reachable states to the initial programs . . . . .	31
5.1	Simple-minded Hoare Logic/VCG for CIMP . . . . .	36
5.1.1	VCG rules . . . . .	39
5.1.2	Cheap non-interference rules . . . . .	41
<b>6</b>	<b>One locale per process</b>	<b>42</b>
<b>7</b>	<b>Example: a one-place buffer</b>	<b>43</b>
<b>8</b>	<b>Example: an unbounded buffer</b>	<b>45</b>
<b>9</b>	<b>Concluding remarks</b>	<b>46</b>
	<b>References</b>	<b>48</b>

## 1 Point-free notation

Typically we define predicates as functions of a state. The following provide a somewhat comfortable point-free imitation of Isabelle/HOL's operators.

**abbreviation** (*input*)

$pred-K :: 'b \Rightarrow 'a \Rightarrow 'b \langle(-)\rangle$  **where**  
 $\langle f \rangle \equiv \lambda s. f$

**abbreviation** (*input*)

$pred-not :: ('a \Rightarrow bool) \Rightarrow 'a \Rightarrow bool \ (\neg - [40] \ 40)$  **where**

$\neg a \equiv \lambda s. \neg a s$

**abbreviation** (*input*)

*pred-conj* :: ('a ⇒ bool) ⇒ ('a ⇒ bool) ⇒ 'a ⇒ bool (**infixr** ∧ 35) **where**  
 $a \wedge b \equiv \lambda s. a s \wedge b s$

**abbreviation** (*input*)

*pred-disj* :: ('a ⇒ bool) ⇒ ('a ⇒ bool) ⇒ 'a ⇒ bool (**infixr** ∨ 30) **where**  
 $a \vee b \equiv \lambda s. a s \vee b s$

**abbreviation** (*input*)

*pred-implies* :: ('a ⇒ bool) ⇒ ('a ⇒ bool) ⇒ 'a ⇒ bool (**infixr** → 25) **where**  
 $a \longrightarrow b \equiv \lambda s. a s \longrightarrow b s$

**abbreviation** (*input*)

*pred-iff* :: ('a ⇒ bool) ⇒ ('a ⇒ bool) ⇒ 'a ⇒ bool (**infixr** ↔ 25) **where**  
 $a \longleftrightarrow b \equiv \lambda s. a s \longleftrightarrow b s$

**abbreviation** (*input*)

*pred-eq* :: ('a ⇒ 'b) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ bool (**infix** = 40) **where**  
 $a = b \equiv \lambda s. a s = b s$

**abbreviation** (*input*)

*pred-member* :: ('a ⇒ 'b) ⇒ ('a ⇒ 'b set) ⇒ 'a ⇒ bool (**infix** ∈ 40) **where**  
 $a \in b \equiv \lambda s. a s \in b s$

**abbreviation** (*input*)

*pred-neq* :: ('a ⇒ 'b) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ bool (**infix** ≠ 40) **where**  
 $a \neq b \equiv \lambda s. a s \neq b s$

**abbreviation** (*input*)

*pred-If* :: ('a ⇒ bool) ⇒ ('a ⇒ 'b) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ 'b ((If (-)/ Then (-)/ Else (-)) [0, 0, 10] 10) **where**  
*If P Then x Else y* ≡ λs. if P s then x s else y s

**abbreviation** (*input*)

*pred-less* :: ('a ⇒ 'b::ord) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ bool (**infix** < 40) **where**  
 $a < b \equiv \lambda s. a s < b s$

**abbreviation** (*input*)

*pred-le* :: ('a ⇒ 'b::ord) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ bool (**infix** ≤ 40) **where**  
 $a \leq b \equiv \lambda s. a s \leq b s$

**abbreviation** (*input*)

*pred-plus* :: ('a ⇒ 'b::plus) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ 'b (**infixl** + 65) **where**  
 $a + b \equiv \lambda s. a s + b s$

**abbreviation** (*input*)

*pred-minus* :: ('a ⇒ 'b::minus) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ 'b (**infixl** - 65) **where**  
 $a - b \equiv \lambda s. a s - b s$

**abbreviation** (*input*)

*fun-fanout* :: ('a ⇒ 'b) ⇒ ('a ⇒ 'c) ⇒ 'a ⇒ 'b × 'c (**infix** ⋈ 35) **where**  
 $f \bowtie g \equiv \lambda x. (f x, g x)$

**abbreviation** (*input*)

*pred-all* :: ('b ⇒ 'a ⇒ bool) ⇒ 'a ⇒ bool (**binder** ∀ 10) **where**  
 $\forall x. P x \equiv \lambda s. \forall x. P x s$

**abbreviation** (*input*)

*pred-ex* :: ('b ⇒ 'a ⇒ bool) ⇒ 'a ⇒ bool (**binder** ∃ 10) **where**  
∃ x. P x ≡ λs. ∃ x. P x s

**abbreviation** (*input*)

*pred-app* :: ('b ⇒ 'a ⇒ 'c) ⇒ ('a ⇒ 'b) ⇒ 'a ⇒ 'c (**infixl** \$ 100) **where**  
f \$ g ≡ λs. f (g s) s

**abbreviation** (*input*)

*pred-subseteq* :: ('a ⇒ 'b set) ⇒ ('a ⇒ 'b set) ⇒ 'a ⇒ bool (**infix** ⊆ 50) **where**  
A ⊆ B ≡ λs. A s ⊆ B s

**abbreviation** (*input*)

*pred-union* :: ('a ⇒ 'b set) ⇒ ('a ⇒ 'b set) ⇒ 'a ⇒ 'b set (**infixl** ∪ 65) **where**  
a ∪ b ≡ λs. a s ∪ b s

**abbreviation** (*input*)

*pred-inter* :: ('a ⇒ 'b set) ⇒ ('a ⇒ 'b set) ⇒ 'a ⇒ 'b set (**infixl** ∩ 65) **where**  
a ∩ b ≡ λs. a s ∩ b s

More application specific.

**abbreviation** (*input*)

*pred-conjoin* :: ('a ⇒ bool) list ⇒ 'a ⇒ bool **where**  
*pred-conjoin* xs ≡ foldr (∧) xs ⟨True⟩

**abbreviation** (*input*)

*pred-disjoin* :: ('a ⇒ bool) list ⇒ 'a ⇒ bool **where**  
*pred-disjoin* xs ≡ foldr (∨) xs ⟨False⟩

**abbreviation** (*input*)

*pred-is-none* :: ('a ⇒ 'b option) ⇒ 'a ⇒ bool (*NULL* - [40] 40) **where**  
*NULL* a ≡ λs. a s = None

**abbreviation** (*input*)

*pred-empty* :: ('a ⇒ 'b set) ⇒ 'a ⇒ bool (*EMPTY* - [40] 40) **where**  
*EMPTY* a ≡ λs. a s = {}

**abbreviation** (*input*)

*pred-list-null* :: ('a ⇒ 'b list) ⇒ 'a ⇒ bool (*LIST-NULL* - [40] 40) **where**  
*LIST-NULL* a ≡ λs. a s = []

**abbreviation** (*input*)

*pred-list-append* :: ('a ⇒ 'b list) ⇒ ('a ⇒ 'b list) ⇒ 'a ⇒ 'b list (**infixr** @ 65) **where**  
xs @ ys ≡ λs. xs s @ ys s

**abbreviation** (*input*)

*pred-pair* :: ('a ⇒ 'b) ⇒ ('a ⇒ 'c) ⇒ 'a ⇒ 'b × 'c (**infixr** ⊗ 60) **where**  
a ⊗ b ≡ λs. (a s, b s)

**abbreviation** (*input*)

*pred-singleton* :: ('a ⇒ 'b) ⇒ 'a ⇒ 'b set **where**  
*pred-singleton* x ≡ λs. {x s}

## 2 Infinite Sequences

Infinite sequences and some operations on them.

We use the customary function-based representation.

**type-synonym**  $'a \text{ seq} = \text{nat} \Rightarrow 'a$   
**type-synonym**  $'a \text{ seq-pred} = 'a \text{ seq} \Rightarrow \text{bool}$

**definition**  $\text{suffix} :: 'a \text{ seq} \Rightarrow \text{nat} \Rightarrow 'a \text{ seq}$  (**infixl**  $|_s$  60) **where**  
 $\sigma |_s i \equiv \lambda j. \sigma (j + i)$

**primrec**  $\text{stake} :: \text{nat} \Rightarrow 'a \text{ seq} \Rightarrow 'a \text{ list}$  **where**  
 $\text{stake } 0 \sigma = []$   
 $| \text{stake } (\text{Suc } n) \sigma = \sigma 0 \# \text{stake } n (\sigma |_s 1)$

**primrec**  $\text{shift} :: 'a \text{ list} \Rightarrow 'a \text{ seq} \Rightarrow 'a \text{ seq}$  (**infixr**  $\langle @- \rangle$  65) **where**  
 $\text{shift } [] \sigma = \sigma$   
 $| \text{shift } (x \# xs) \sigma = (\lambda i. \text{case } i \text{ of } 0 \Rightarrow x | \text{Suc } i \Rightarrow \text{shift } xs \sigma i)$

**abbreviation**  $\text{interval-syn} (-'(- \rightarrow -')$  [69, 0, 0] 70) **where**  
 $\sigma(i \rightarrow j) \equiv \text{stake } j (\sigma |_s i)$

**lemma**  $\text{suffix-eval}$ :  $(\sigma |_s i) j = \sigma (j + i)$   
**unfolding**  $\text{suffix-def}$  **by**  $\text{simp}$

**lemma**  $\text{suffix-plus}$ :  $\sigma |_s n |_s m = \sigma |_s (m + n)$   
**unfolding**  $\text{suffix-def}$  **by**  $(\text{simp add: add.assoc})$

**lemma**  $\text{suffix-commute}$ :  $((\sigma |_s n) |_s m) = ((\sigma |_s m) |_s n)$   
**by**  $(\text{simp add: suffix-plus add.commute})$

**lemma**  $\text{suffix-plus-com}$ :  $\sigma |_s m |_s n = \sigma |_s (m + n)$

**proof** –

**have**  $\sigma |_s n |_s m = \sigma |_s (m + n)$  **by**  $(\text{rule suffix-plus})$

**then show**  $\sigma |_s m |_s n = \sigma |_s (m + n)$  **by**  $(\text{simp add: suffix-commute})$

**qed**

**lemma**  $\text{suffix-zero}$ :  $\sigma |_s 0 = \sigma$

**unfolding**  $\text{suffix-def}$  **by**  $\text{simp}$

**lemma**  $\text{comp-suffix}$ :  $f \circ \sigma |_s i = (f \circ \sigma) |_s i$

**unfolding**  $\text{suffix-def comp-def}$  **by**  $\text{simp}$

**lemmas**  $\text{suffix-simps}[\text{simp}] =$

$\text{comp-suffix}$   
 $\text{suffix-eval}$   
 $\text{suffix-plus-com}$   
 $\text{suffix-zero}$

**lemma**  $\text{length-stake}[\text{simp}]$ :  $\text{length } (\text{stake } n s) = n$

**by**  $(\text{induct } n \text{ arbitrary: } s) \text{ auto}$

**lemma**  $\text{shift-simps}[\text{simp}]$ :

$(xs @- \sigma) 0 = (\text{if } xs = [] \text{ then } \sigma 0 \text{ else } \text{hd } xs)$

$(xs @- \sigma) |_s \text{Suc } 0 = (\text{if } xs = [] \text{ then } \sigma |_s \text{Suc } 0 \text{ else } \text{tl } xs @- \sigma)$

**by**  $(\text{induct } xs) \text{ auto}$

**lemma**  $\text{stake-nil}[\text{simp}]$ :

$\text{stake } i \sigma = [] \longleftrightarrow i = 0$

**by**  $(\text{cases } i; \text{clarsimp})$

**lemma**  $\text{stake-shift}$ :

$stake\ i\ (w\ @-\ \sigma) = take\ i\ w\ @\ stake\ (i - length\ w)\ \sigma$   
**by** (*induct i arbitrary: w*) (*auto simp: neq-Nil-conv*)

**lemma** *shift-snth-less*[*simp*]:

**assumes**  $i < length\ xs$   
**shows**  $(xs\ @-\ \sigma)\ i = xs\ !\ i$

**using** *assms*

**proof**(*induct i arbitrary: xs*)

**case** (*Suc i xs*) **then show** *?case* **by** (*cases xs*) *simp-all*  
**qed** (*simp add: hd-conv-nth nth-tl*)

**lemma** *shift-snth-ge*[*simp*]:

**assumes**  $i \geq length\ xs$   
**shows**  $(xs\ @-\ \sigma)\ i = \sigma\ (i - length\ xs)$

**using** *assms*

**proof**(*induct i arbitrary: xs*)

**case** (*Suc i xs*) **then show** *?case* **by** (*cases xs*) *simp-all*  
**qed** *simp*

**lemma** *shift-snth*:

$(xs\ @-\ \sigma)\ i = (if\ i < length\ xs\ then\ xs\ !\ i\ else\ \sigma\ (i - length\ xs))$

**by** *simp*

**lemma** *suffix-shift*:

$(xs\ @-\ \sigma)\ |_s\ i = drop\ i\ xs\ @-\ (\sigma\ |_s\ i - length\ xs)$

**proof**(*induct i arbitrary: xs*)

**case** (*Suc i xs*) **then show** *?case* **by** (*cases xs*) *simp-all*  
**qed** *simp*

**lemma** *stake-nth*[*simp*]:

**assumes**  $i < j$   
**shows**  $stake\ j\ s\ !\ i = s\ i$

**using** *assms* **by** (*induct j arbitrary: s i*) (*simp-all add: nth-Cons'*)

**lemma** *stake-suffix-id*:

$stake\ i\ \sigma\ @-\ (\sigma\ |_s\ i) = \sigma$

**by** (*induct i*) (*simp-all add: fun-eq-iff shift-snth split: nat.splits*)

**lemma** *id-stake-snth-suffix*:

$\sigma = (stake\ i\ \sigma\ @\ [\sigma\ i])\ @-\ (\sigma\ |_s\ Suc\ i)$

**using** *stake-suffix-id*

**apply** (*metis Suc-diff-le append-Nil2 diff-is-0-eq length-stake lessI nat.simps(3) nat-le-linear shift-snth stake-nil stake-shift take-Suc-conv-app-nth*)

**done**

**lemma** *stake-add*[*simp*]:

$stake\ i\ \sigma\ @\ stake\ j\ (\sigma\ |_s\ i) = stake\ (i + j)\ \sigma$

**apply** (*induct i arbitrary: sigma*)

**apply** *simp*

**apply** *auto*

**apply** (*metis One-nat-def plus-1-eq-Suc suffix-plus-com*)

**done**

**lemma** *stake-append*:  $stake\ n\ (u\ @-\ s) = take\ (min\ (length\ u)\ n)\ u\ @\ stake\ (n - length\ u)\ s$

**proof** (*induct n arbitrary: u*)

**case** (*Suc n*) **then show** *?case*

**apply** *clarsimp*

**apply** (*cases u*)

```

apply auto
done
qed auto

```

```

lemma stake-shift-stake-shift:
  stake i  $\sigma$  @- stake j ( $\sigma$  |s i) @-  $\beta$  = stake (i + j)  $\sigma$  @-  $\beta$ 
apply (induct i arbitrary:  $\sigma$ )
apply simp
apply auto
apply (metis One-nat-def plus-1-eq-Suc suffix-plus-com)
done

```

```

lemma stake-suffix-drop:
  stake i ( $\sigma$  |s j) = drop j (stake (i + j)  $\sigma$ )
by (metis append-eq-conv-conj length-stake semiring-normalization-rules(24) stake-add)

```

```

lemma stake-suffix:
  assumes  $i \leq j$ 
  shows stake j  $\sigma$  @-  $u$  |s i =  $\sigma(i \rightarrow j - i)$  @-  $u$ 
by (simp add: asms stake-suffix-drop suffix-shift)

```

## 2.1 Decomposing safety and liveness

Famously properties on infinite sequences can be decomposed into *safety* and *liveness* properties [Alpern and Schneider \(1985\)](#); [Schneider \(1987\)](#). See [Kindler \(1994\)](#) for an overview.

```

definition safety :: 'a seq-pred  $\Rightarrow$  bool where
  safety P  $\longleftrightarrow$  ( $\forall \sigma. \neg P \sigma \longrightarrow (\exists i. \forall \beta. \neg P (\text{stake } i \sigma @- \beta))$ )

```

```

lemma safety-def2: — Contraposition gives the customary prefix-closure definition
  safety P  $\longleftrightarrow$  ( $\forall \sigma. (\forall i. \exists \beta. P (\text{stake } i \sigma @- \beta)) \longrightarrow P \sigma$ )
unfolding safety-def by blast

```

```

definition liveness :: 'a seq-pred  $\Rightarrow$  bool where
  liveness P  $\longleftrightarrow$  ( $\forall \alpha. \exists \sigma. P (\alpha @- \sigma)$ )

```

```

lemmas safetyI = iffD2[OF safety-def, rule-format]
lemmas safetyI2 = iffD2[OF safety-def2, rule-format]
lemmas livenessI = iffD2[OF liveness-def, rule-format]

```

```

lemma safety-False:
  shows safety ( $\lambda \sigma. \text{False}$ )
by (rule safetyI) simp

```

```

lemma safety-True:
  shows safety ( $\lambda \sigma. \text{True}$ )
by (rule safetyI) simp

```

```

lemma safety-state-prop:
  shows safety ( $\lambda \sigma. P (\sigma 0)$ )
by (rule safetyI) auto

```

```

lemma safety-invariant:
  shows safety ( $\lambda \sigma. \forall i. P (\sigma i)$ )
apply (rule safetyI)
apply clarsimp
apply (metis length-stake lessI shift-snth-less stake-nth)
done

```

**lemma** *safety-transition-relation*:

**shows** *safety*  $(\lambda\sigma. \forall i. (\sigma\ i, \sigma\ (i + 1)) \in R)$

**apply** (*rule safetyI*)

**apply** *clarsimp*

**apply** (*metis* (*no-types, hide-lams*) *Suc-eq-plus1 add.left-neutral add-Suc-right add-diff-cancel-left' le-add1 list.sel(1) list.simps(3) shift-simps(1) stake.simps(2) stake-suffix suffix-def*)

**done**

**lemma** *safety-conj*:

**assumes** *safety P*

**assumes** *safety Q*

**shows** *safety*  $(P \wedge Q)$

**using** *assms unfolding safety-def by blast*

**lemma** *safety-always-eventually[simplified]*:

**assumes** *safety P*

**assumes**  $\forall i. \exists j \geq i. \exists \beta. P\ (\sigma(0 \rightarrow j)\ @- \beta)$

**shows** *P*  $\sigma$

**using** *assms unfolding safety-def2*

**apply** –

**apply** (*drule-tac*  $x=\sigma$  **in** *spec*)

**apply** *clarsimp*

**apply** (*drule-tac*  $x=i$  **in** *spec*)

**apply** *clarsimp*

**apply** (*rule-tac*  $x=(stake\ j\ \sigma\ @- \beta)\ |_s\ i$  **in** *exI*)

**apply** (*simp add: stake-shift-stake-shift stake-suffix*)

**done**

**lemma** *safety-disj*:

**assumes** *safety P*

**assumes** *safety Q*

**shows** *safety*  $(P \vee Q)$

**unfolding** *safety-def2 using assms*

**by** (*metis safety-always-eventually add-diff-cancel-right' diff-le-self le-add-same-cancel2*)

The decomposition is given by a form of closure.

**definition**  $M_p :: 'a\ seq-pred \Rightarrow 'a\ seq-pred$  **where**

$M_p\ P = (\lambda\sigma. \forall i. \exists \beta. P\ (stake\ i\ \sigma\ @- \beta))$

**definition**  $Safe :: 'a\ seq-pred \Rightarrow 'a\ seq-pred$  **where**

$Safe\ P = (P \vee M_p\ P)$

**definition**  $Live :: 'a\ seq-pred \Rightarrow 'a\ seq-pred$  **where**

$Live\ P = (P \vee \neg M_p\ P)$

**lemma** *decomp*:

$P = (Safe\ P \wedge Live\ P)$

**unfolding** *Safe-def Live-def by blast*

**lemma** *safe*:

*safety*  $(Safe\ P)$

**unfolding** *Safe-def safety-def M<sub>p</sub>-def*

**apply** *clarsimp*

**apply** (*simp add: stake-shift*)

**apply** (*rule-tac*  $x=i$  **in** *exI*)

**apply** *clarsimp*

**apply** (*rule-tac*  $x=i$  **in** *exI*)

**apply** *clarsimp*

done

**lemma** *live*:

*liveness* (*Live P*)

**proof**(*rule livenessI*)

**fix**  $\alpha$

**have**  $(\exists \beta. P (\alpha @- \beta)) \vee \neg(\exists \beta. P (\alpha @- \beta))$  **by** *blast*

**also have**  $?this \longleftrightarrow (\exists \beta. P (\alpha @- \beta) \vee (\forall \gamma. \neg P (\alpha @- \gamma)))$  **by** *blast*

**also have**  $\dots \longleftrightarrow (\exists \beta. P (\alpha @- \beta) \vee (\exists i. i = \text{length } \alpha \wedge (\forall \gamma. \neg P (\text{stake } i (\alpha @- \beta) @- \gamma))))$  **by** (*simp add: stake-shift*)

**also have**  $\dots \longrightarrow (\exists \beta. P (\alpha @- \beta) \vee (\exists i. (\forall \gamma. \neg P (\text{stake } i (\alpha @- \beta) @- \gamma))))$  **by** *blast*

**finally have**  $\exists \beta. P (\alpha @- \beta) \vee (\exists i. \forall \gamma. \neg P (\text{stake } i (\alpha @- \beta) @- \gamma))$  .

**then show**  $\exists \sigma. \text{Live } P (\alpha @- \sigma)$  **unfolding** *Live-def M<sub>p</sub>-def* **by** *simp*

**qed**

Sistla (1994) proceeds to give a topological analysis of fairness. An *absolute* liveness property is a liveness property whose complement is stable.

**definition** *absolute-liveness* :: '*a seq-pred*  $\Rightarrow$  *bool* **where** — closed under prepending any finite sequence  
*absolute-liveness*  $P \longleftrightarrow (\exists \sigma. P \sigma) \wedge (\forall \sigma \alpha. P \sigma \longrightarrow P (\alpha @- \sigma))$

**definition** *stable* :: '*a seq-pred*  $\Rightarrow$  *bool* **where** — closed under suffixes  
*stable*  $P \longleftrightarrow (\exists \sigma. P \sigma) \wedge (\forall \sigma i. P \sigma \longrightarrow P (\sigma |_s i))$

**lemma** *absolute-liveness-liveness*:

**assumes** *absolute-liveness P*

**shows** *liveness P*

**using** *assms* **unfolding** *absolute-liveness-def liveness-def* **by** *blast*

**lemma** *stable-absolute-liveness*:

**assumes**  $P \sigma$

**assumes**  $\neg P \sigma'$  — extra hypothesis

**shows** *stable P*  $\longleftrightarrow$  *absolute-liveness* ( $\neg P$ )

**using** *assms* **unfolding** *stable-def absolute-liveness-def*

**apply** *auto*

**apply** (*metis cancel-comm-monoid-add-class.diff-cancel drop-eq-Nil order-refl shift.simps*(1) *suffix-shift suffix-zero*)

**apply** (*metis stake-suffix-id*)

**done**

**definition** *fairness* :: '*a seq-pred*  $\Rightarrow$  *bool* **where**  
*fairness P*  $\longleftrightarrow$  *stable P*  $\wedge$  *absolute-liveness P*

**lemma** *fairness-safety*:

**assumes** *safety P*

**assumes** *fairness F*

**shows**  $(\forall \sigma. F \sigma \longrightarrow P \sigma) \longleftrightarrow (\forall \sigma. P \sigma)$

**apply** *rule*

**using** *assms*

**apply** *clarsimp*

**unfolding** *safety-def fairness-def stable-def absolute-liveness-def*

**apply** *clarsimp*

**apply** *blast+*

**done**



### 3 Linear Temporal Logic

To talk about liveness we need to consider infinitary behaviour on sequences. Traditionally future-time linear temporal logic (LTL) is used to do this [Manna and Pnueli \(1991\)](#); [Owicki and Lamport \(1982\)](#).

The following is a straightforward shallow embedding of the now-traditional anchored semantics of LTL [Manna and Pnueli \(1988\)](#). Some of it is adapted from the sophisticated TLA development in the AFP due to [Grover and Merz \(2011\)](#).

Unlike [Lamport \(2002\)](#), include the next operator, which is convenient for stating rules. Sometimes it allows us to ignore the system, i.e. to state rules as temporally valid (LTL-valid) rather than just temporally program valid (LTL-cimp-), in Jackson's terminology.

**definition** *state-prop* :: ('a ⇒ bool) ⇒ 'a seq-pred ([·]) **where**  
 $[P] = (\lambda\sigma. P (\sigma 0))$

**definition** *next* :: 'a seq-pred ⇒ 'a seq-pred (○- [80] 80) **where**  
 $(\circ P) = (\lambda\sigma. P (\sigma |_s 1))$

**definition** *always* :: 'a seq-pred ⇒ 'a seq-pred (□- [80] 80) **where**  
 $(\square P) = (\lambda\sigma. \forall i. P (\sigma |_s i))$

**definition** *until* :: 'a seq-pred ⇒ 'a seq-pred ⇒ 'a seq-pred (**infixr**  $\mathcal{U}$  30) **where**  
 $(P \mathcal{U} Q) = (\lambda\sigma. \exists i. Q (\sigma |_s i) \wedge (\forall k < i. P (\sigma |_s k)))$

**definition** *eventually* :: 'a seq-pred ⇒ 'a seq-pred (◇- [80] 80) **where**  
 $(\diamond P) = (\langle \text{True} \rangle \mathcal{U} P)$

**definition** *release* :: 'a seq-pred ⇒ 'a seq-pred ⇒ 'a seq-pred (**infixr**  $\mathcal{R}$  30) **where**  
 $(P \mathcal{R} Q) = (\neg(\neg P \mathcal{U} \neg Q))$

**definition** *unless* :: 'a seq-pred ⇒ 'a seq-pred ⇒ 'a seq-pred (**infixr**  $\mathcal{W}$  30) **where**  
 $(P \mathcal{W} Q) = ((P \mathcal{U} Q) \vee \square P)$

**abbreviation** (*input*)

*pred-always-imp-syn* :: 'a seq-pred ⇒ 'a seq-pred ⇒ 'a seq-pred (**infixr**  $\hookrightarrow$  25) **where**  
 $P \hookrightarrow Q \equiv \square(P \longrightarrow Q)$

**lemmas** *defs* =

*state-prop-def*  
*always-def*  
*eventually-def*  
*next-def*  
*release-def*  
*unless-def*  
*until-def*

**lemma** *suffix-state-prop[simp]*:

**shows**  $[P] (\sigma |_s i) = P (\sigma i)$

**unfolding** *defs* **by** *simp*

**lemma** *alwaysI[intro]*:

**assumes**  $\bigwedge i. P (\sigma |_s i)$

**shows**  $(\square P) \sigma$

**unfolding** *defs* **using** *assms* **by** *blast*

**lemma** *alwaysD*:

**assumes**  $(\square P) \sigma$

**shows**  $P (\sigma |_s i)$

**using** *assms* **unfolding** *defs* **by** *blast*

**lemma** *alwaysE*:  $\llbracket (\Box P) \sigma; P (\sigma \mid_s i) \implies Q \rrbracket \implies Q$   
**unfolding** *defs by blast*

**lemma** *always-induct*:

**assumes**  $P \sigma$   
**assumes**  $(\Box(P \longrightarrow \circ P)) \sigma$   
**shows**  $(\Box P) \sigma$

**proof**(*rule alwaysI*)

**fix**  $i$  **from** *assms* **show**  $P (\sigma \mid_s i)$   
**unfolding** *defs by (induct i) simp-all*

**qed**

**lemma** *seq-comp*:

**fixes**  $\sigma :: 'a \text{ seq}$   
**fixes**  $P :: 'b \text{ seq-pred}$   
**fixes**  $f :: 'a \Rightarrow 'b$   
**shows**

$(\Box P) (f \circ \sigma) \longleftrightarrow (\Box(\lambda\sigma. P (f \circ \sigma))) \sigma$   
 $(\Diamond P) (f \circ \sigma) \longleftrightarrow (\Diamond(\lambda\sigma. P (f \circ \sigma))) \sigma$   
 $(P \mathcal{U} Q) (f \circ \sigma) \longleftrightarrow ((\lambda\sigma. P (f \circ \sigma)) \mathcal{U} (\lambda\sigma. Q (f \circ \sigma))) \sigma$   
 $(P \mathcal{W} Q) (f \circ \sigma) \longleftrightarrow ((\lambda\sigma. P (f \circ \sigma)) \mathcal{W} (\lambda\sigma. Q (f \circ \sigma))) \sigma$

**unfolding** *defs by simp-all*

**lemma** *nextI[intro]*:

**assumes**  $P (\sigma \mid_s \text{Suc } 0)$   
**shows**  $(\circ P) \sigma$

**using** *assms unfolding defs by simp*

**lemma** *untilI[intro]*:

**assumes**  $Q (\sigma \mid_s i)$   
**assumes**  $\forall k < i. P (\sigma \mid_s k)$   
**shows**  $(P \mathcal{U} Q) \sigma$

**unfolding** *defs using assms by blast*

**lemma** *untilE*:

**assumes**  $(P \mathcal{U} Q) \sigma$   
**obtains**  $i$  **where**  $Q (\sigma \mid_s i)$  **and**  $\forall k < i. P (\sigma \mid_s k)$

**using** *assms unfolding until-def by blast*

**lemma** *eventuallyI[intro]*:

**assumes**  $P (\sigma \mid_s i)$   
**shows**  $(\Diamond P) \sigma$

**unfolding** *eventually-def using assms by blast*

**lemma** *eventuallyE[elim]*:

**assumes**  $(\Diamond P) \sigma$   
**obtains**  $i$  **where**  $P (\sigma \mid_s i)$

**using** *assms unfolding defs by (blast elim: untilE)*

**lemma** *unless-alwaysI*:

**assumes**  $(\Box P) \sigma$   
**shows**  $(P \mathcal{W} Q) \sigma$

**using** *assms unfolding defs by blast*

**lemma** *unless-untilI*:

**assumes**  $Q (\sigma \mid_s j)$   
**assumes**  $\bigwedge i. i < j \implies P (\sigma \mid_s i)$

shows  $(P \mathcal{W} Q) \sigma$   
**unfolding** *defs* **using** *assms* **by** *blast*

**lemma** *always-imp-refl[iff]*:  
 shows  $(P \leftrightarrow P) \sigma$   
**unfolding** *defs* **by** *blast*

**lemma** *always-imp-trans*:  
 assumes  $(P \leftrightarrow Q) \sigma$   
 assumes  $(Q \leftrightarrow R) \sigma$   
 shows  $(P \leftrightarrow R) \sigma$   
**using** *assms* **unfolding** *defs* **by** *blast*

**lemma** *always-imp-mp*:  
 assumes  $(P \leftrightarrow Q) \sigma$   
 assumes  $P \sigma$   
 shows  $Q \sigma$   
**using** *assms* **unfolding** *defs* **by** (*metis suffix-zero*)

**lemma** *always-imp-mp-suffix*:  
 assumes  $(P \leftrightarrow Q) \sigma$   
 assumes  $P (\sigma \mid_s i)$   
 shows  $Q (\sigma \mid_s i)$   
**using** *assms* **unfolding** *defs* **by** *metis*

Some basic facts and equivalences, mostly sanity.

**lemma** *necessitation*:  
 $(\bigwedge s. P s) \implies (\Box P) \sigma$   
 $(\bigwedge s. P s) \implies (\Diamond P) \sigma$   
 $(\bigwedge s. P s) \implies (P \mathcal{W} Q) \sigma$   
 $(\bigwedge s. Q s) \implies (P \mathcal{U} Q) \sigma$   
**unfolding** *defs* **by** *auto*

**lemma** *cong*:  
 $(\bigwedge s. P s = P' s) \implies [P] = [P']$   
 $(\bigwedge \sigma. P \sigma = P' \sigma) \implies (\Box P) = (\Box P')$   
 $(\bigwedge \sigma. P \sigma = P' \sigma) \implies (\Diamond P) = (\Diamond P')$   
 $(\bigwedge \sigma. P \sigma = P' \sigma) \implies (\circ P) = (\circ P')$   
 $\llbracket \bigwedge \sigma. P \sigma = P' \sigma; \bigwedge \sigma. Q \sigma = Q' \sigma \rrbracket \implies (P \mathcal{U} Q) = (P' \mathcal{U} Q')$   
 $\llbracket \bigwedge \sigma. P \sigma = P' \sigma; \bigwedge \sigma. Q \sigma = Q' \sigma \rrbracket \implies (P \mathcal{W} Q) = (P' \mathcal{W} Q')$   
**unfolding** *defs* **by** *auto*

**lemma** *norm[simp]*:  
 $[\langle False \rangle] = \langle False \rangle$   
 $[\langle True \rangle] = \langle True \rangle$   
 $(\neg [p]) = [\neg p]$   
 $([p] \wedge [q]) = [p \wedge q]$   
 $([p] \vee [q]) = [p \vee q]$   
 $([p] \longrightarrow [q]) = [p \longrightarrow q]$   
 $([p] \sigma \wedge [q] \sigma) = [p \wedge q] \sigma$   
 $([p] \sigma \vee [q] \sigma) = [p \vee q] \sigma$   
 $([p] \sigma \longrightarrow [q] \sigma) = [p \longrightarrow q] \sigma$

$(\circ \langle False \rangle) = \langle False \rangle$   
 $(\circ \langle True \rangle) = \langle True \rangle$

$(\Box \langle False \rangle) = \langle False \rangle$   
 $(\Box \langle True \rangle) = \langle True \rangle$

$$(\neg \Box P) \sigma = (\Diamond (\neg P)) \sigma$$

$$(\Box \Box P) = (\Box P)$$

$$(\Diamond \langle \text{False} \rangle) = \langle \text{False} \rangle$$

$$(\Diamond \langle \text{True} \rangle) = \langle \text{True} \rangle$$

$$(\neg \Diamond P) = (\Box (\neg P))$$

$$(\Diamond \Diamond P) = (\Diamond P)$$

$$(P \mathcal{W} \langle \text{False} \rangle) = (\Box P)$$

$$(\neg(P \mathcal{U} Q)) \sigma = (\neg P \mathcal{R} \neg Q) \sigma$$

$$(\langle \text{False} \rangle \mathcal{U} P) = P$$

$$(P \mathcal{U} \langle \text{False} \rangle) = \langle \text{False} \rangle$$

$$(P \mathcal{U} \langle \text{True} \rangle) = \langle \text{True} \rangle$$

$$(\langle \text{True} \rangle \mathcal{U} P) = (\Diamond P)$$

$$(P \mathcal{U} (P \mathcal{U} Q)) = (P \mathcal{U} Q)$$

$$(\neg(P \mathcal{R} Q)) \sigma = (\neg P \mathcal{U} \neg Q) \sigma$$

$$(\langle \text{False} \rangle \mathcal{R} P) = (\Box P)$$

$$(P \mathcal{R} \langle \text{False} \rangle) = \langle \text{False} \rangle$$

$$(\langle \text{True} \rangle \mathcal{R} P) = P$$

$$(P \mathcal{R} \langle \text{True} \rangle) = \langle \text{True} \rangle$$

**unfolding defs**

**apply** (*auto simp: fun-eq-iff*)

**apply** (*metis suffix-plus suffix-zero*)

**apply** (*metis suffix-plus suffix-zero*)

**apply** *fastforce*

**apply** *force*

**apply** (*metis add.commute add-diff-inverse-nat less-diff-conv2 not-le*)

**apply** (*metis add.right-neutral not-less0*)

**apply** *force*

**apply** *fastforce*

**done**

**lemma** *always-conj-distrib*:  $(\Box(P \wedge Q)) = (\Box P \wedge \Box Q)$

**unfolding defs by auto**

**lemma** *eventually-disj-distrib*:  $(\Diamond(P \vee Q)) = (\Diamond P \vee \Diamond Q)$

**unfolding defs by auto**

**lemma** *always-eventually[elim!]*:

**assumes**  $(\Box P) \sigma$

**shows**  $(\Diamond P) \sigma$

**using** *assms unfolding defs by auto*

**lemma** *eventually-imp-conv-disj*:  $(\Diamond(P \longrightarrow Q)) = (\Diamond(\neg P) \vee \Diamond Q)$

**unfolding defs by auto**

**lemma** *eventually-imp-distrib*:

$(\Diamond(P \longrightarrow Q)) = (\Box P \longrightarrow \Diamond Q)$

**unfolding defs by auto**

**lemma** *unfold*:

$(\Box P) \sigma = (P \wedge \circ \Box P) \sigma$

$(\Diamond P) \sigma = (P \vee \circ \Diamond P) \sigma$

$(P \mathcal{W} Q) \sigma = (Q \vee (P \wedge \circ (P \mathcal{W} Q))) \sigma$

$(P \mathcal{U} Q) \sigma = (Q \vee (P \wedge \circ (P \mathcal{U} Q))) \sigma$

$(P \mathcal{R} Q) \sigma = (Q \wedge (P \vee \circ(P \mathcal{R} Q))) \sigma$   
**unfolding** *defs*  
**apply** –  
**apply** (*metis* (*full-types*) *add.commute add-diff-inverse-nat less-one suffix-plus suffix-zero*)  
**apply** (*metis* (*full-types*) *One-nat-def add.right-neutral add-Suc-right lessI less-Suc-eq-0-disj suffix-plus suffix-zero*)  
**apply** *auto*  
**apply** *fastforce*  
**apply** (*metis* *gr0-conv-Suc nat-neq-iff not-less-eq suffix-zero*)  
**apply** (*metis* *suffix-zero*)  
**apply** *force*  
**using** *less-Suc-eq-0-disj* **apply** *fastforce*  
**apply** (*metis* *gr0-conv-Suc nat-neq-iff not-less0 suffix-zero*)  
  
**apply** *fastforce*  
**apply** (*case-tac* *i*; *auto*)  
**apply** *force*  
**using** *less-Suc-eq-0-disj* **apply** *force*  
  
**apply** *force*  
**using** *less-Suc-eq-0-disj* **apply** *fastforce*  
**apply** *fastforce*  
**apply** (*case-tac* *i*; *auto*)  
**done**  
  
**lemma** *mono*:  
 $\llbracket (\Box P) \sigma; \bigwedge \sigma. P \sigma \implies P' \sigma \rrbracket \implies (\Box P') \sigma$   
 $\llbracket (\Diamond P) \sigma; \bigwedge \sigma. P \sigma \implies P' \sigma \rrbracket \implies (\Diamond P') \sigma$   
 $\llbracket (P \mathcal{U} Q) \sigma; \bigwedge \sigma. P \sigma \implies P' \sigma; \bigwedge \sigma. Q \sigma \implies Q' \sigma \rrbracket \implies (P' \mathcal{U} Q') \sigma$   
 $\llbracket (P \mathcal{W} Q) \sigma; \bigwedge \sigma. P \sigma \implies P' \sigma; \bigwedge \sigma. Q \sigma \implies Q' \sigma \rrbracket \implies (P' \mathcal{W} Q') \sigma$   
**unfolding** *defs* **by** *force+*  
  
**lemma** *always-imp-mono*:  
 $\llbracket (\Box P) \sigma; (P \hookrightarrow P') \sigma \rrbracket \implies (\Box P') \sigma$   
 $\llbracket (\Diamond P) \sigma; (P \hookrightarrow P') \sigma \rrbracket \implies (\Diamond P') \sigma$   
 $\llbracket (P \mathcal{U} Q) \sigma; (P \hookrightarrow P') \sigma; (Q \hookrightarrow Q') \sigma \rrbracket \implies (P' \mathcal{U} Q') \sigma$   
 $\llbracket (P \mathcal{W} Q) \sigma; (P \hookrightarrow P') \sigma; (Q \hookrightarrow Q') \sigma \rrbracket \implies (P' \mathcal{W} Q') \sigma$   
**unfolding** *defs* **by** *force+*  
  
**lemma** *next-conj-distrib*:  
 $(\circ(P \wedge Q)) = (\circ P \wedge \circ Q)$   
**unfolding** *defs* **by** *auto*  
  
**lemma** *next-disj-distrib*:  
 $(\circ(P \vee Q)) = (\circ P \vee \circ Q)$   
**unfolding** *defs* **by** *auto*  
  
**lemma** *until-next-distrib*:  
 $(\circ(P \mathcal{U} Q)) = (\circ P \mathcal{U} \circ Q)$   
**unfolding** *defs* **by** (*auto simp: fun-eq-iff*)  
  
**lemma** *until-imp-eventually*:  
 $((P \mathcal{U} Q) \longrightarrow \Diamond Q) \sigma$   
**unfolding** *defs* **by** *auto*  
  
**lemma** *until-until-disj*:  
**assumes**  $(P \mathcal{U} Q \mathcal{U} R) \sigma$   
**shows**  $((P \vee Q) \mathcal{U} R) \sigma$

**using** *assms unfolding defs*  
**apply** *clarsimp*  
**apply** (*metis (full-types) add-diff-inverse-nat nat-add-left-cancel-less*)  
**done**

**lemma** *unless-unless-disj*:  
**assumes**  $(P \ \mathcal{W} \ Q \ \mathcal{W} \ R) \ \sigma$   
**shows**  $((P \vee Q) \ \mathcal{W} \ R) \ \sigma$   
**using** *assms unfolding defs*  
**apply** *auto*  
**apply** (*metis add.commute add-diff-inverse-nat leI less-diff-conv2*)  
**apply** (*metis add-diff-inverse-nat*)  
**done**

**lemma** *until-conj-distrib*:  
 $((P \wedge Q) \ \mathcal{U} \ R) = ((P \ \mathcal{U} \ R) \wedge (Q \ \mathcal{U} \ R))$   
**unfolding** *defs*  
**apply** (*auto simp: fun-eq-iff*)  
**apply** (*metis dual-order.strict-trans nat-neq-iff*)  
**done**

**lemma** *until-disj-distrib*:  
 $(P \ \mathcal{U} \ (Q \vee R)) = ((P \ \mathcal{U} \ Q) \vee (P \ \mathcal{U} \ R))$   
**unfolding** *defs* **by** (*auto simp: fun-eq-iff*)

**lemma** *eventually-until*:  
 $(\diamond P) = (\neg P \ \mathcal{U} \ P)$   
**unfolding** *defs*  
**apply** (*auto simp: fun-eq-iff*)  
**apply** (*case-tac P (x |s 0)*)  
**apply** *blast*  
**apply** (*drule (1) ex-least-nat-less*)  
**apply** (*metis le-simps(2)*)  
**done**

**lemma** *eventually-until-eventually*:  
 $(\diamond(P \ \mathcal{U} \ Q)) = (\diamond Q)$   
**unfolding** *defs* **by** *force*

**lemma** *eventually-unless-until*:  
 $((P \ \mathcal{W} \ Q) \wedge \diamond Q) = (P \ \mathcal{U} \ Q)$   
**unfolding** *defs* **by** *force*

**lemma** *eventually-always-imp-always-eventually*:  
**assumes**  $(\diamond \square P) \ \sigma$   
**shows**  $(\square \diamond P) \ \sigma$   
**using** *assms unfolding defs* **by** (*metis suffix-commute*)

**lemma** *eventually-always-next-stable*:  
**assumes**  $(\diamond P) \ \sigma$   
**assumes**  $(P \ \leftrightarrow \ \circ P) \ \sigma$   
**shows**  $(\diamond \square P) \ \sigma$   
**using** *assms* **by** (*metis (no-types) eventuallyI alwaysD always-induct eventuallyE norm(15)*)

**lemma** *next-stable-imp-eventually-always*:  
**assumes**  $(P \ \leftrightarrow \ \circ P) \ \sigma$   
**shows**  $(\diamond P \ \longrightarrow \ \diamond \square P) \ \sigma$   
**using** *assms* **eventually-always-next-stable** **by** *blast*

**lemma** *always-eventually-always*:

$$\diamond \square \diamond P = \square \diamond P$$

**unfolding** *defs* by (*clarsimp simp: fun-eq-iff*) (*metis add.left-commute semiring-normalization-rules(25)*)

**lemma** *stable-unless*:

**assumes**  $(P \leftrightarrow \circ(P \vee Q)) \sigma$

**shows**  $(P \leftrightarrow (P \mathcal{W} Q)) \sigma$

**using** *assms unfolding defs*

**apply**  $-$

**apply** (*rule ccontr*)

**apply** *clarsimp*

**apply** (*drule (1) ex-least-nat-less[where P= $\lambda j. \neg P (\sigma \mid_s i + j)$  for  $i$ , simplified]*)

**apply** *clarsimp*

**apply** (*metis add-Suc-right le-less less-Suc-eq*)

**done**

**lemma** *unless-induct*: — Rule WAIT from [Manna and Pnueli \(1995, Fig 3.3\)](#)

**assumes**  $I: (I \leftrightarrow \circ(I \vee R)) \sigma$

**assumes**  $P: (P \leftrightarrow I \vee R) \sigma$

**assumes**  $Q: (I \leftrightarrow Q) \sigma$

**shows**  $(P \leftrightarrow Q \mathcal{W} R) \sigma$

**apply** (*intro alwaysI impI*)

**apply** (*erule impE[OF alwaysD[OF P]]*)

**apply** (*erule disjE*)

**apply** (*rule always-imp-mono(4)[where P= $I$  and  $Q= $R$ ]$* )

**apply** (*erule mp[OF alwaysD[OF stable-unless[OF I]]]*)

**apply** (*simp add: Q alwaysD*)

**apply** *blast*

**apply** (*simp add: unfold*)

**done**

### 3.1 Leads-to and leads-to-via

Most of our assertions will be of the form  $\lambda s. A s \longrightarrow (\diamond C) s$  (pronounced “ $A$  leads to  $C$ ”) or  $\lambda s. A s \longrightarrow (B \mathcal{U} C) s$  (“ $A$  leads to  $C$  via  $B$ ”).

Most of these rules are due to [Jackson \(1998\)](#) who used leads-to-via in a sequential setting. Others are due to [Manna and Pnueli \(1991\)](#).

The leads-to-via connective is similar to the “ensures” modality of [Chandy and Misra \(1989, §3.4.4\)](#).

**abbreviation** (*input*)

*leads-to* :: ' $a$  seq-pred  $\Rightarrow$  ' $a$  seq-pred  $\Rightarrow$  ' $a$  seq-pred (**infixr**  $\rightsquigarrow$  25) **where**

$$P \rightsquigarrow Q \equiv P \leftrightarrow \diamond Q$$

**lemma** *leads-to-refl*:

**shows**  $(P \rightsquigarrow P) \sigma$

**by** (*metis (no-types, lifting) necessitation(1) unfold(2)*)

**lemma** *leads-to-trans*:

**assumes**  $(P \rightsquigarrow Q) \sigma$

**assumes**  $(Q \rightsquigarrow R) \sigma$

**shows**  $(P \rightsquigarrow R) \sigma$

**using** *assms unfolding defs* by *clarsimp (metis semiring-normalization-rules(25))*

**lemma** *leads-to-eventuallyE*:

**assumes**  $(P \rightsquigarrow Q) \sigma$

**assumes**  $(\diamond P) \sigma$

**shows**  $(\diamond Q) \sigma$   
**using** *assms unfolding defs by auto*

**lemma** *leads-to-mono*:  
**assumes**  $(P' \hookrightarrow P) \sigma$   
**assumes**  $(Q \hookrightarrow Q') \sigma$   
**assumes**  $(P \rightsquigarrow Q) \sigma$   
**shows**  $(P' \rightsquigarrow Q') \sigma$   
**using** *assms unfolding defs by clarsimp blast*

**lemma** *leads-to-eventually*:  
**shows**  $(P \rightsquigarrow Q \longrightarrow \diamond P \longrightarrow \diamond Q) \sigma$   
**by** (*metis (no-types, lifting) alwaysI unfold(2)*)

**lemma** *leads-to-disj*:  
**assumes**  $(P \rightsquigarrow R) \sigma$   
**assumes**  $(Q \rightsquigarrow R) \sigma$   
**shows**  $((P \vee Q) \rightsquigarrow R) \sigma$   
**using** *assms unfolding defs by simp*

**lemma** *leads-to-leads-to-viaE*:  
**shows**  $((P \hookrightarrow P \mathcal{U} Q) \longrightarrow P \rightsquigarrow Q) \sigma$   
**unfolding** *defs by clarsimp blast*

**lemma** *leads-to-via-concl-weaken*:  
**assumes**  $(R \hookrightarrow R') \sigma$   
**assumes**  $(P \hookrightarrow Q \mathcal{U} R) \sigma$   
**shows**  $(P \hookrightarrow Q \mathcal{U} R') \sigma$   
**using** *assms unfolding LTL.defs by force*

**lemma** *leads-to-via-trans*:  
**assumes**  $(A \hookrightarrow B \mathcal{U} C) \sigma$   
**assumes**  $(C \hookrightarrow D \mathcal{U} E) \sigma$   
**shows**  $(A \hookrightarrow (B \vee D) \mathcal{U} E) \sigma$   
**proof**(*rule alwaysI, rule impI*)  
**fix** *i* **assume**  $A (\sigma \mid_s i)$  **with** *assms* **show**  $((B \vee D) \mathcal{U} E) (\sigma \mid_s i)$   
**apply** –  
**apply** (*erule alwaysE[where i=i]*)  
**apply** *clarsimp*  
**apply** (*erule untilE*)  
**apply** *clarsimp*  
**apply** (*drule (1) always-imp-mp-suffix*)  
**apply** (*erule untilE*)  
**apply** *clarsimp*  
**apply** (*rule-tac i=ia + iaa in untilI; simp add: ac-simps*)  
**apply** (*metis (full-types) add.assoc leI le-Suc-ex nat-add-left-cancel-less*)  
**done**  
**qed**

**lemma** *leads-to-via-disj*: — useful for case distinctions  
**assumes**  $(P \hookrightarrow Q \mathcal{U} R) \sigma$   
**assumes**  $(P' \hookrightarrow Q' \mathcal{U} R) \sigma$   
**shows**  $(P \vee P' \hookrightarrow (Q \vee Q') \mathcal{U} R) \sigma$   
**using** *assms unfolding defs by (auto 10 0)*

**lemma** *leads-to-via-disj'*: — more like a chaining rule  
**assumes**  $(A \hookrightarrow B \mathcal{U} C) \sigma$   
**assumes**  $(C \hookrightarrow D \mathcal{U} E) \sigma$



```

shows  $(A \vee C \hookrightarrow (B \vee D) \mathcal{U} E) \sigma$ 
proof(rule alwaysI, rule impI, erule disjE)
  fix i assume  $A (\sigma \upharpoonright_s i)$  with assms show  $((B \vee D) \mathcal{U} E) (\sigma \upharpoonright_s i)$ 
    apply -
    apply (erule alwaysE[where i=i])
    apply clarsimp
    apply (erule untilE)
    apply clarsimp
    apply (drule (1) always-imp-mp-suffix)
    apply (erule untilE)
    apply clarsimp
    apply (rule-tac i=ia + iaa in untilI; simp add: ac-simps)
    apply (metis (full-types) add.assoc leI le-Suc-ex nat-add-left-cancel-less)
  done
next
  fix i assume  $C (\sigma \upharpoonright_s i)$  with assms(2) show  $((B \vee D) \mathcal{U} E) (\sigma \upharpoonright_s i)$ 
    apply -
    apply (erule alwaysE[where i=i])
    apply (simp add: mono)
  done
qed

lemma leads-to-via-stable-augmentation:
  assumes stable:  $(P \wedge Q \hookrightarrow \circ Q) \sigma$ 
  assumes U:  $(A \hookrightarrow P \mathcal{U} C) \sigma$ 
  shows  $((A \wedge Q) \hookrightarrow P \mathcal{U} (C \wedge Q)) \sigma$ 
proof(intro alwaysI impI, elim conjE)
  fix i assume AP:  $A (\sigma \upharpoonright_s i) Q (\sigma \upharpoonright_s i)$ 
  have  $Q (\sigma \upharpoonright_s (j + i))$  if  $Q (\sigma \upharpoonright_s i)$  and  $\forall k < j. P (\sigma \upharpoonright_s (k + i))$  for j
    using that stable by (induct j; force simp: defs)
  with U AP show  $(P \mathcal{U} (\lambda\sigma. C \sigma \wedge Q \sigma)) (\sigma \upharpoonright_s i)$ 
    unfolding defs by clarsimp (metis (full-types) add commute)
qed

lemma leads-to-via-wf:
  assumes wf R
  assumes indhyp:  $\bigwedge t. (A \wedge [\delta = \langle t \rangle] \hookrightarrow B \mathcal{U} (A \wedge [\delta \otimes \langle t \rangle \in \langle R \rangle] \vee C)) \sigma$ 
  shows  $(A \hookrightarrow B \mathcal{U} C) \sigma$ 
proof(intro alwaysI impI)
  fix i assume  $A (\sigma \upharpoonright_s i)$  with wf R show  $(B \mathcal{U} C) (\sigma \upharpoonright_s i)$ 
  proof(induct  $\delta (\sigma \upharpoonright_s i)$  arbitrary: i)
    case (less i) with indhyp[where  $t=\delta (\sigma \upharpoonright_s i)$ ] show ?case
      apply -
      apply (drule alwaysD[where i=i])
      apply clarsimp
      apply (erule untilE)
      apply (rename-tac j)
      apply (erule disjE; clarsimp)
      apply (drule-tac  $x=i + j$  in meta-spec; clarsimp)
      apply (erule untilE; clarsimp)
      apply (rename-tac j k)
      apply (rule-tac  $i=j + k$  in untilI)
      apply (simp add: add.assoc)
      apply clarsimp
      apply (metis add.assoc add commute add-diff-inverse-nat less-diff-conv2 not-le)
  apply auto
  done
qed

```

qed

The well-founded response rule due to [Manna and Pnueli \(2010, Fig 1.23: WELL \(well-founded response\)\)](#), generalised to an arbitrary set of assertions and sequence predicates.

- $W1$  generalised to be contingent.
- $W2$  is a well-founded set of assertions that by  $W1$  includes  $P$

**lemma** *leads-to-wf*:

**fixes**  $Is :: ('a \text{ seq-pred} \times ('a \Rightarrow 'b)) \text{ set}$

**assumes**  $wf (R :: 'b \text{ rel})$

**assumes**  $W1: (\Box(\exists \varphi. [\langle \varphi \in fst \ 'Is \rangle] \wedge (P \longrightarrow \varphi))) \sigma$

**assumes**  $W2: \forall (\varphi, \delta) \in Is. \exists (\varphi', \delta') \in insert (Q, \delta 0) Is. \forall t. (\varphi \wedge [\delta = \langle t \rangle] \rightsquigarrow \varphi' \wedge [\delta' \otimes \langle t \rangle \in \langle R \rangle]) \sigma$

**shows**  $(P \rightsquigarrow Q) \sigma$

**proof** –

**have**  $(\varphi \wedge [\delta = \langle t \rangle] \rightsquigarrow Q) \sigma$  **if**  $(\varphi, \delta) \in Is$  **for**  $\varphi \ \delta \ t$

**using**  $\langle wf \ R \rangle$  **that**  $W2$

**apply**  $(induct \ t \ \text{arbitrary}: \varphi \ \delta)$

**unfolding**  $LTL.defs \ split-def$

**apply**  $clarsimp$

**apply**  $(metis (no-types, hide-lams) ab-semigroup-add-class.add-ac(1) fst-eqD snd-conv surjective-pairing)$

**done**

**with**  $W1$  **show**  $?thesis$

**apply** –

**apply**  $(rule \ alwaysI)$

**apply**  $clarsimp$

**apply**  $(erule-tac \ i=i \ \text{in} \ alwaysE)$

**apply**  $clarsimp$

**using**  $alwaysD \ suffix-state-prop$  **apply**  $blast$

**done**

qed

## 3.2 Fairness

A few renderings of weak fairness. [van Glabbeek and Höfner \(2019\)](#) call this "response to insistence" as a generalisation of weak fairness.

**definition** *weakly-fair*  $:: 'a \text{ seq-pred} \Rightarrow 'a \text{ seq-pred} \Rightarrow 'a \text{ seq-pred}$  **where**

$weakly-fair \ enabled \ taken = (\Box enabled \leftrightarrow \Diamond taken)$

**lemma** *weakly-fair-def2*:

**shows**  $weakly-fair \ enabled \ taken = \Box(\neg \Box(enabled \wedge \neg taken))$

**unfolding** *weakly-fair-def* **by**  $(metis (full-types) always-conj-distrib norm(18))$

**lemma** *weakly-fair-def3*:

**shows**  $weakly-fair \ enabled \ taken = (\Diamond \Box enabled \longrightarrow \Box \Diamond taken)$

**unfolding** *weakly-fair-def2*

**apply**  $(clarsimp \ simp: fun-eq-iff)$

**unfolding** *defs*

**apply** *auto*

**apply**  $(metis (full-types) add.left-commute semiring-normalization-rules(25))$

**done**

**lemma** *weakly-fair-def4*:

**shows**  $weakly-fair \ enabled \ taken = \Box \Diamond (enabled \longrightarrow taken)$

**using** *weakly-fair-def2* **by** *force*

**lemma** *mp-weakly-fair*:

**assumes** *weakly-fair enabled taken*  $\sigma$

**assumes**  $(\Box \text{enabled}) \sigma$

**shows**  $(\Diamond \text{taken}) \sigma$

**using** *assms unfolding weakly-fair-def using always-imp-mp by blast*

**lemma** *always-weakly-fair*:

**shows**  $\Box(\text{weakly-fair enabled taken}) = \text{weakly-fair enabled taken}$

**unfolding** *weakly-fair-def by simp*

**lemma** *eventually-weakly-fair*:

**shows**  $\Diamond(\text{weakly-fair enabled taken}) = \text{weakly-fair enabled taken}$

**unfolding** *weakly-fair-def2 by (simp add: always-eventually-always)*

**lemma** *weakly-fair-weaken*:

**assumes**  $(\text{enabled}' \leftrightarrow \text{enabled}) \sigma$

**assumes**  $(\text{taken} \leftrightarrow \text{taken}') \sigma$

**shows**  $(\text{weakly-fair enabled taken} \leftrightarrow \text{weakly-fair enabled}' \text{ taken}') \sigma$

**using** *assms unfolding weakly-fair-def defs by simp blast*

**lemma** *weakly-fair-unless-until*:

**shows**  $(\text{weakly-fair enabled taken} \wedge (\text{enabled} \leftrightarrow \text{enabled } \mathcal{W} \text{ taken})) = (\text{enabled} \leftrightarrow \text{enabled } \mathcal{U} \text{ taken})$

**unfolding** *defs weakly-fair-def*

**apply** *(auto simp: fun-eq-iff)*

**apply** *(metis add.right-neutral)*

**done**

**lemma** *stable-leads-to-eventually*:

**assumes**  $(\text{enabled} \leftrightarrow \bigcirc(\text{enabled} \vee \text{taken})) \sigma$

**shows**  $(\text{enabled} \leftrightarrow (\Box \text{enabled} \vee \Diamond \text{taken})) \sigma$

**using** *assms unfolding defs*

**apply**  $-$

**apply** *(rule ccontr)*

**apply** *clarsimp*

**apply** *(drule (1) ex-least-nat-less[where  $P=\lambda j. \neg \text{enabled} (\sigma \upharpoonright_s i + j)$  for  $i$ , simplified])*

**apply** *clarsimp*

**apply** *(metis add-Suc-right leI less-irrefl-nat)*

**done**

**lemma** *weakly-fair-stable-leads-to*:

**assumes**  $(\text{weakly-fair enabled taken}) \sigma$

**assumes**  $(\text{enabled} \leftrightarrow \bigcirc(\text{enabled} \vee \text{taken})) \sigma$

**shows**  $(\text{enabled} \rightsquigarrow \text{taken}) \sigma$

**using** *stable-leads-to-eventually[OF assms(2)] assms(1) unfolding defs weakly-fair-def*

**by** *(auto simp: fun-eq-iff)*

**lemma** *weakly-fair-stable-leads-to-via*:

**assumes**  $(\text{weakly-fair enabled taken}) \sigma$

**assumes**  $(\text{enabled} \leftrightarrow \bigcirc(\text{enabled} \vee \text{taken})) \sigma$

**shows**  $(\text{enabled} \leftrightarrow \text{enabled } \mathcal{U} \text{ taken}) \sigma$

**using** *stable-unless[OF assms(2)] assms(1) by (metis (mono-tags) weakly-fair-unless-until)*

Similarly for strong fairness. van Glabbeek and Höfner (2019) call this "response to persistence" as a generalisation of strong fairness.

**definition** *strongly-fair*  $:: 'a \text{ seq-pred} \Rightarrow 'a \text{ seq-pred} \Rightarrow 'a \text{ seq-pred}$  **where**

*strongly-fair enabled taken*  $= (\Box \Diamond \text{enabled} \leftrightarrow \Diamond \text{taken})$

**lemma** *strongly-fair-def2*:

*strongly-fair enabled taken* =  $\Box(\neg\Box(\Diamond\text{enabled} \wedge \neg\text{taken}))$   
**unfolding** *strongly-fair-def* **by** (*metis weakly-fair-def weakly-fair-def2*)

**lemma** *strongly-fair-def3*:

*strongly-fair enabled taken* =  $(\Box\Diamond\text{enabled} \longrightarrow \Box\Diamond\text{taken})$

**unfolding** *strongly-fair-def2* **by** (*metis (full-types) always-eventually-always weakly-fair-def2 weakly-fair-def3*)

**lemma** *always-strongly-fair*:

$\Box(\text{strongly-fair enabled taken}) = \text{strongly-fair enabled taken}$

**unfolding** *strongly-fair-def* **by** *simp*

**lemma** *eventually-strongly-fair*:

$\Diamond(\text{strongly-fair enabled taken}) = \text{strongly-fair enabled taken}$

**unfolding** *strongly-fair-def2* **by** (*simp add: always-eventually-always*)

**lemma** *strongly-fair-disj-distrib*: — not true for *weakly-fair*

*strongly-fair (enabled1  $\vee$  enabled2) taken* = (*strongly-fair enabled1 taken*  $\wedge$  *strongly-fair enabled2 taken*)

**unfolding** *strongly-fair-def defs*

**apply** (*auto simp: fun-eq-iff*)

**apply** *blast*

**apply** *blast*

**apply** (*metis (full-types) semiring-normalization-rules(25)*)

**done**

**lemma** *strongly-fair-imp-weakly-fair*:

**assumes** *strongly-fair enabled taken*  $\sigma$

**shows** *weakly-fair enabled taken*  $\sigma$

**using** *assms* **unfolding** *strongly-fair-def3 weakly-fair-def3* **by** (*simp add: eventually-always-imp-always-eventually*)

**lemma** *always-enabled-weakly-fair-strongly-fair*:

**assumes**  $(\Box\text{enabled}) \sigma$

**shows** *weakly-fair enabled taken*  $\sigma = \text{strongly-fair enabled taken}$   $\sigma$

**using** *assms* **by** (*metis strongly-fair-def3 strongly-fair-imp-weakly-fair unfold(2) weakly-fair-def3*)

### 3.3 Safety and liveness

Sistla (1994) shows some characterisations of LTL formulas in terms of safety and liveness. Note his ( $\mathcal{U}$ ) is actually ( $\mathcal{W}$ ).

See also Chang, Manna, and Pnueli (1992).

**lemma** *safety-state-prop*:

**shows** *safety*  $\lceil P \rceil$

**unfolding** *defs* **by** (*rule safety-state-prop*)

**lemma** *safety-Next*:

**assumes** *safety*  $P$

**shows** *safety*  $(\bigcirc P)$

**using** *assms* **unfolding** *defs safety-def*

**apply** *clarsimp*

**apply** (*metis (mono-tags) One-nat-def list.sel(3) nat.simps(3) stake.simps(2)*)

**done**

**lemma** *safety-unless*:

**assumes** *safety*  $P$

**assumes** *safety*  $Q$

**shows** *safety*  $(P \mathcal{W} Q)$

**proof**(*rule safetyI2*)

**fix**  $\sigma$  **assume**  $X: \exists \beta. (P \mathcal{W} Q) (\text{stake } i \sigma @- \beta)$  **for**  $i$

```

then show (P W Q)  $\sigma$ 
proof(cases  $\forall i j. \exists \beta. P (\sigma(i \rightarrow j) @- \beta)$ )
  case True
  with  $\langle \text{safety } P \rangle$  have  $\forall i. P (\sigma |_s i)$  unfolding safety-def2 by blast
  then show ?thesis by (blast intro: unless-alwaysI)
next
  case False
  then obtain  $k k'$  where  $\forall \beta. \neg P (\sigma(k \rightarrow k') @- \beta)$  by clarsimp
  then have  $\forall i u. k + k' \leq i \longrightarrow \neg P ((\text{stake } i \sigma @- u) |_s k)$ 
    by (metis add commute diff-add stake-shift-stake-shift stake-suffix-drop suffix-shift)
  then have  $\forall i u. k + k' \leq i \wedge (P W Q) (\text{stake } i \sigma @- u) \longrightarrow (\exists m \leq k. Q ((\text{stake } i \sigma @- u) |_s m) \wedge (\forall p < m. P ((\text{stake } i \sigma @- u) |_s p)))$ 
    unfolding defs using leI by blast
  then have  $\forall i u. k + k' \leq i \wedge (P W Q) (\text{stake } i \sigma @- u) \longrightarrow (\exists m \leq k. Q (\sigma(m \rightarrow i - m) @- u) \wedge (\forall p < m. P (\sigma(p \rightarrow i - p) @- u)))$ 
    by (metis stake-suffix add-leE nat-less-le order-trans)
  then have  $\forall i. \exists n \geq i. \exists m \leq k. \exists u. Q (\sigma(m \rightarrow n - m) @- u) \wedge (\forall p < m. P (\sigma(p \rightarrow n - p) @- u))$ 
    using X by (metis add commute le-add1)
  then have  $\exists m \leq k. \forall i. \exists n \geq i. \exists u. Q (\sigma(m \rightarrow n - m) @- u) \wedge (\forall p < m. P (\sigma(p \rightarrow n - p) @- u))$ 
    by (simp add: always-eventually-pigeonhole)
  with  $\langle \text{safety } P \rangle \langle \text{safety } Q \rangle$  show (P W Q)  $\sigma$ 
    unfolding defs by (metis Nat.le-diff-conv2 add-leE safety-always-eventually)
qed
qed

```

```

lemma safety-always:
  assumes safety P
  shows safety ( $\square P$ )
using assms by (metis norm(20) safety-def safety-unless)

```

```

lemma absolute-liveness-eventually:
  shows absolute-liveness P  $\longleftrightarrow (\exists \sigma. P \sigma) \wedge P = \diamond P$ 
unfolding absolute-liveness-def defs
by (metis cancel-comm-monoid-add-class.diff-cancel drop-eq-Nil order-refl shift.simps(1) stake-suffix-id suffix-shift suffix-zero)

```

```

lemma stable-always:
  shows stable P  $\longleftrightarrow (\exists \sigma. P \sigma) \wedge P = \square P$ 
unfolding stable-def defs by (metis suffix-zero)

```

To show that *weakly-fair* is a *fairness* property requires some constraints on *enabled* and *taken*:

- it is reasonable to assume they are state formulas
- *taken* must be satisfiable

```

lemma fairness-weakly-fair:
  assumes  $\exists s. \textit{taken } s$ 
  shows fairness (weakly-fair [enabled] [taken])
unfolding fairness-def stable-def absolute-liveness-def weakly-fair-def
using assms
apply auto
  apply (rule-tac x= $\lambda$ . .s in exI)
  apply fastforce
  apply (simp add: alwaysD)
  apply (rule-tac x= $\lambda$ . .s in exI)
  apply fastforce
apply (metis (full-types) absolute-liveness-def absolute-liveness-eventually eventually-weakly-fair weakly-fair-def)
done

```

```

lemma fairness-strongly-fair:
  assumes  $\exists s. \text{taken } s$ 
  shows fairness (strongly-fair [enabled] [taken])
unfolding fairness-def stable-def absolute-liveness-def strongly-fair-def
using assms
apply auto
  apply (rule-tac  $x=\lambda-.s$  in exI)
  apply fastforce
  apply (simp add: alwaysD)
  apply (rule-tac  $x=\lambda-.s$  in exI)
  apply fastforce
apply (metis (full-types) absolute-liveness-def absolute-liveness-eventually eventually-weakly-fair weakly-fair-def)
done

```

## 4 CIMP syntax and semantics

We define a small sequential programming language with synchronous message passing primitives for describing the individual processes. This has the advantage over raw transition systems in that it is programmer-readable, includes sequential composition, supports a program logic and VCG (§5.1), etc. These processes are composed in parallel at the top-level.

CIMP is inspired by IMP, as presented by Winskel (1993) and Nipkow and Klein (2014), and the classical process algebras CCS (Milner 1980, 1989) and CSP (Hoare 1985). Note that the algebraic properties of this language have not been developed.

As we operate in a concurrent setting, we need to provide a small-step semantics (§4.2), which we give in the style of *structural operational semantics* (SOS) as popularised by Plotkin (2004). The semantics of a complete system (§4.3) is presently taken simply to be the states reachable by interleaving the enabled steps of the individual processes, subject to message passing rendezvous. We leave a trace or branching semantics to future work.

This theory contains all the trusted definitions. The soundness of the other theories supervenes upon this one.

### 4.1 Syntax

Programs are represented using an explicit (deep embedding) of their syntax, as the semantics needs to track the progress of multiple threads of control. Each (atomic) *basic command* (§??) is annotated with a *'location*, which we use in our assertions (§4.4). These locations need not be unique, though in practice they likely will be.

Processes maintain *local states* of type *'state*. These can be updated with arbitrary relations of *'state*  $\Rightarrow$  *'state set* with *LocalOp*, and conditions of type *'s*  $\Rightarrow$  *bool* are similarly shallowly embedded. This arrangement allows the end-user to select their own level of atomicity.

The sequential composition operator and control constructs are standard. We add the infinite looping construct *Loop* so we can construct single-state reactive systems; this has implications for fairness assertions.

**type-synonym** *'s bexp* = *'s*  $\Rightarrow$  *bool*

```

datatype ('answer, 'location, 'question, 'state) com
  = Request 'location 'state  $\Rightarrow$  'question 'answer  $\Rightarrow$  'state  $\Rightarrow$  'state set      ( $\{\}$ -} Request - - [0, 70, 70] 71)
  | Response 'location 'question  $\Rightarrow$  'state  $\Rightarrow$  ('state  $\times$  'answer) set      ( $\{\}$ -} Response - [0, 70] 71)
  | LocalOp 'location 'state  $\Rightarrow$  'state set                                ( $\{\}$ -} LocalOp - [0, 70] 71)
  | Cond1   'location 'state bexp ('answer, 'location, 'question, 'state) com ( $\{\}$ -} IF - THEN - FI [0, 0, 0] 71)
  | Cond2   'location 'state bexp ('answer, 'location, 'question, 'state) com
              ('answer, 'location, 'question, 'state) com          ( $\{\}$ -} IF -/ THEN -/ ELSE -/ FI [0, 0,
0, 0] 71)
  | Loop    ('answer, 'location, 'question, 'state) com                (LOOP DO -/ OD [0] 71)
  | While   'location 'state bexp ('answer, 'location, 'question, 'state) com ( $\{\}$ -} WHILE -/ DO -/ OD [0, 0, 0] 71)
  | Seq     ('answer, 'location, 'question, 'state) com
              ('answer, 'location, 'question, 'state) com          (infixr ;; 69)
  | Choose ('answer, 'location, 'question, 'state) com

```

We provide a one-armed conditional as it is the common form and avoids the need to discover a label for an internal *SKIP* and/or trickier proofs about the VCG.

In contrast to classical process algebras, we have local state and distinct request and response actions. These provide an interface to Isabelle/HOL's datatypes that avoids the need for binding (ala the  $\pi$ -calculus of Milner (1989)) or large non-deterministic sums (ala CCS (Milner 1980, §2.8)). Intuitively the requester poses a *'question* with a *Request* command, which upon rendezvous with a responder's *Response* command receives an *'answer*. The *'question* is a deterministic function of the requester's local state, whereas responses can be non-deterministic. Note that CIMP does not provide a notion of channel; these can be modelled by a judicious choice of *'question*.

We also provide a binary external choice operator ( $\oplus$ ) (infix ( $\oplus$ )). Internal choice can be recovered in combination with local operations (see Milner (1980, §2.3)).

We abbreviate some common commands: *SKIP* is a local operation that does nothing, and the floor brackets simplify deterministic *LocalOps*. We also adopt some syntax magic from Makarius's *Hoare* and *Multiquote* theories in the Isabelle/HOL distribution.

**abbreviation** *SKIP-syn* ( $\{\!-\!\}$ / *SKIP* [0] 71) **where**

$\{\!l\!\} \text{ SKIP} \equiv \{\!l\!\} \text{ LocalOp } (\lambda s. \{s\})$

**abbreviation** (*input*) *DetLocalOp* :: *'location*  $\Rightarrow$  (*'state*  $\Rightarrow$  *'state*)

$\Rightarrow$  (*'answer, 'location, 'question, 'state*) com ( $\{\!-\!\}$  [-] [0, 0] 71) **where**

$\{\!l\!\} [f] \equiv \{\!l\!\} \text{ LocalOp } (\lambda s. \{f s\})$

**syntax**

-*quote* :: *'b*  $\Rightarrow$  (*'a*  $\Rightarrow$  *'b*) ( $\ll$ - $\gg$  [0] 1000)

-*antiquote* :: (*'a*  $\Rightarrow$  *'b*)  $\Rightarrow$  *'b* (- [1000] 1000)

-*Assign* :: *'location*  $\Rightarrow$  *idt*  $\Rightarrow$  *'b*  $\Rightarrow$  (*'answer, 'location, 'question, 'state*) com (( $\{\!-\!\}$ ' - :=/ -) [0, 0, 70] 71)

-*NonDetAssign* :: *'location*  $\Rightarrow$  *idt*  $\Rightarrow$  *'b set*  $\Rightarrow$  (*'answer, 'location, 'question, 'state*) com (( $\{\!-\!\}$ ' - :=/ -) [0, 0, 70] 71)

**abbreviation** (*input*) *NonDetAssign* :: *'location*  $\Rightarrow$  ((*'val*  $\Rightarrow$  *'val*)  $\Rightarrow$  *'state*  $\Rightarrow$  *'state*)  $\Rightarrow$  (*'state*  $\Rightarrow$  *'val set*)

$\Rightarrow$  (*'answer, 'location, 'question, 'state*) com **where**

*NonDetAssign l upd es*  $\equiv \{\!l\!\} \text{ LocalOp } (\lambda s. \{ \text{upd } \langle e \rangle s \mid e. e \in es \ s \})$

**translations**

$\{\!l\!\}' x := e \Rightarrow \text{CONST } \text{DetLocalOp } l \ll(-\text{update-name } x (\lambda-. e))\gg$

$\{\!l\!\}' x \in es \Rightarrow \text{CONST } \text{NonDetAssign } l (-\text{update-name } x) \ll es \gg$

**parse-translation** (

*let*

*fun antiquote-tr i* (*Const* (@{*syntax-const antiquote*}, -) \$  
 (*t as Const* (@{*syntax-const antiquote*}, -) \$ -) = *skip-antiquote-tr i t*  
 | *antiquote-tr i* (*Const* (@{*syntax-const antiquote*}, -) \$ *t*) =  
*antiquote-tr i t* \$ *Bound i*  
 | *antiquote-tr i* (*t* \$ *u*) = *antiquote-tr i t* \$ *antiquote-tr i u*  
 | *antiquote-tr i* (*Abs* (*x*, *T*, *t*)) = *Abs* (*x*, *T*, *antiquote-tr (i + 1) t*)  
 | *antiquote-tr - a* = *a*  
*and skip-antiquote-tr i* ((*c as Const* (@{*syntax-const antiquote*}, -) \$ *t*) =  
*c* \$ *skip-antiquote-tr i t*  
 | *skip-antiquote-tr i t* = *antiquote-tr i t*;

*fun quote-tr [t]* = *Abs* (*s*, *dummyT*, *antiquote-tr 0 (Term.incr-boundvars 1 t)*)

| *quote-tr ts* = *raise TERM (quote-tr, ts)*;

*in* [(@{*syntax-const quote*}, *K quote-tr*)] *end*

)

## 4.2 Process semantics

Here we define the semantics of a single process's program. We begin by defining the type of externally-visible behaviour:

**datatype** ('answer, 'question) seq-label  
 = sl-Internal ( $\tau$ )  
 | sl-Send 'question 'answer («-, -»)   
 | sl-Receive 'question 'answer (»-, -«)

We define a *labelled transition system* (an LTS) using an execution-stack style of semantics that avoids special treatment of the *SKIPS* introduced by a traditional small step semantics (such as Winskel (1993, Chapter 14)) when a basic command is executed. This was suggested by Thomas Sewell; Pitts (2002) gave a semantics to an ML-like language using this approach.

We record the location of the command that was executed to support fairness constraints.

**type-synonym** ('answer, 'location, 'question, 'state) local-state  
 = ('answer, 'location, 'question, 'state) com list  $\times$  'location option  $\times$  'state

### inductive

*small-step* :: ('answer, 'location, 'question, 'state) local-state  
 $\Rightarrow$  ('answer, 'question) seq-label  
 $\Rightarrow$  ('answer, 'location, 'question, 'state) local-state  $\Rightarrow$  bool (-  $\rightarrow$  . - [55, 0, 56] 55)

### where

$\llbracket \alpha = \text{action } s; s' \in \text{val } \beta \ s \rrbracket \Longrightarrow (\{l\} \text{ Request action val } \# \text{ cs, -, s}) \rightarrow_{\llbracket \alpha, \beta \rrbracket} (\text{cs, Some } l, s')$   
 $\llbracket (s', \beta) \in \text{action } \alpha \ s \rrbracket \Longrightarrow (\{l\} \text{ Response action } \# \text{ cs, -, s}) \rightarrow_{\llbracket \alpha, \beta \llbracket \rrbracket} (\text{cs, Some } l, s')$

$| s' \in R \ s \Longrightarrow (\{l\} \text{ LocalOp } R \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (\text{cs, Some } l, s')$

$| b \ s \Longrightarrow (\{l\} \text{ IF } b \ \text{THEN } c \ \text{FI} \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (c \ \# \ \text{cs, Some } l, s)$

$| \neg b \ s \Longrightarrow (\{l\} \text{ IF } b \ \text{THEN } c \ \text{FI} \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (\text{cs, Some } l, s)$

$| b \ s \Longrightarrow (\{l\} \text{ IF } b \ \text{THEN } c1 \ \text{ELSE } c2 \ \text{FI} \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (c1 \ \# \ \text{cs, Some } l, s)$

$| \neg b \ s \Longrightarrow (\{l\} \text{ IF } b \ \text{THEN } c1 \ \text{ELSE } c2 \ \text{FI} \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (c2 \ \# \ \text{cs, Some } l, s)$

$| (c \ \# \ \text{LOOP DO } c \ \text{OD} \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s') \Longrightarrow (\text{LOOP DO } c \ \text{OD} \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s')$

$| b \ s \Longrightarrow (\{l\} \text{ WHILE } b \ \text{DO } c \ \text{OD} \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (c \ \# \ \{l\} \text{ WHILE } b \ \text{DO } c \ \text{OD} \ \# \ \text{cs, Some } l, s)$

$| \neg b \ s \Longrightarrow (\{l\} \text{ WHILE } b \ \text{DO } c \ \text{OD} \ \# \ \text{cs, -, s}) \rightarrow_{\tau} (\text{cs, Some } l, s)$

$| (c1 \ \# \ c2 \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s') \Longrightarrow (c1;; c2 \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s')$

$| \text{Choose1: } (c1 \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s') \Longrightarrow (c1 \oplus c2 \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s')$

$| \text{Choose2: } (c2 \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s') \Longrightarrow (c1 \oplus c2 \ \# \ \text{cs, s}) \rightarrow_{\alpha} (\text{cs}', s')$

The following projections operate on local states. These should not appear to the end-user.

**abbreviation** *cPGM* :: ('answer, 'location, 'question, 'state) local-state  $\Rightarrow$  ('answer, 'location, 'question, 'state) com list **where**  
*cPGM*  $\equiv$  *fst*

**abbreviation** *cTKN* :: ('answer, 'location, 'question, 'state) local-state  $\Rightarrow$  'location option **where**  
*cTKN*  $s \equiv$  *fst* (*snd* *s*)

**abbreviation** *cLST* :: ('answer, 'location, 'question, 'state) local-state  $\Rightarrow$  'state **where**  
*cLST*  $s \equiv$  *snd* (*snd* *s*)

## 4.3 System steps

A global state maps process names to process' local states. One might hope to allow processes to have distinct types of local state, but there remains no good solution yet in a simply-typed setting; see Schirmer and Wenzel



(2009).

**type-synonym**  $(\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ global-state}$   
 $= \text{'proc} \Rightarrow (\text{'answer}, \text{'location}, \text{'question}, \text{'state}) \text{ local-state}$

**type-synonym**  $(\text{'proc}, \text{'state}) \text{ local-states}$   
 $= \text{'proc} \Rightarrow \text{'state}$

An execution step of the overall system is either any enabled internal  $\tau$  step of any process, or a communication rendezvous between two processes. For the latter to occur, a *Request* action must be enabled in process  $p1$ , and a *Response* action in (distinct) process  $p2$ , where the request/response labels  $\alpha$  and  $\beta$  (semantically) match.

We also track global communication history here to support assertional reasoning (see §5).

**type-synonym**  $(\text{'answer}, \text{'question}) \text{ event} = \text{'question} \times \text{'answer}$   
**type-synonym**  $(\text{'answer}, \text{'question}) \text{ history} = (\text{'answer}, \text{'question}) \text{ event list}$

**record**  $(\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state} =$   
 $GST :: (\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ global-state}$   
 $HST :: (\text{'answer}, \text{'question}) \text{ history}$

**inductive** — This is a predicate of the current state, so the successor state comes first.

$\text{system-step} :: \text{'proc set}$   
 $\Rightarrow (\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state}$   
 $\Rightarrow (\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state}$   
 $\Rightarrow \text{bool}$

**where**

$\text{LocalStep}: \llbracket GST \text{ sh } p \rightarrow_{\tau} \text{ls}' ; GST \text{ sh}' = (GST \text{ sh})(p := \text{ls}'); HST \text{ sh}' = HST \text{ sh} \rrbracket \Longrightarrow \text{system-step } \{p\} \text{ sh}' \text{ sh}$   
 $\mid \text{CommunicationStep}: \llbracket GST \text{ sh } p \rightarrow_{\langle\alpha, \beta\rangle} \text{ls1}' ; GST \text{ sh } q \rightarrow_{\langle\alpha, \beta\rangle} \text{ls2}' ; p \neq q ;$   
 $GST \text{ sh}' = (GST \text{ sh})(p := \text{ls1}', q := \text{ls2}'); HST \text{ sh}' = HST \text{ sh } @ [(\alpha, \beta)] \rrbracket \Longrightarrow \text{system-step}$   
 $\{p, q\} \text{ sh}' \text{ sh}$

In classical process algebras matching communication actions yield  $\tau$  steps, which aids nested parallel composition and the restriction operation (Milner 1980, §2.2). As CIMP does not provide either we do not need to hide communication labels. In CCS/CSP it is not clear how one reasons about the communication history, and it seems that assertional reasoning about these languages is not well developed.

We define predicates over communication histories and system states. These are uncurried to ease composition.

**type-synonym**  $(\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ state-pred}$   
 $= (\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state} \Rightarrow \text{bool}$

The *LST* operator (written as a postfix  $\downarrow$ ) projects the local states of the processes from a  $(\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state}$ , i.e. it discards control location information.

Conversely the *LSTP* operator lifts predicates over local states into predicates over  $(\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state}$ .

Predicates that do not depend on control locations were termed *universal assertions* by Levin and Gries (1981, §3.6).

**type-synonym**  $(\text{'proc}, \text{'state}) \text{ local-state-pred}$   
 $= (\text{'proc}, \text{'state}) \text{ local-states} \Rightarrow \text{bool}$

**definition**  $LST :: (\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ system-state}$   
 $\Rightarrow (\text{'proc}, \text{'state}) \text{ local-states} (-\downarrow [1000] 1000) \text{ where}$   
 $s\downarrow = cLST \circ GST s$

**abbreviation**  $(\text{input}) LSTP :: (\text{'proc}, \text{'state}) \text{ local-state-pred}$   
 $\Rightarrow (\text{'answer}, \text{'location}, \text{'proc}, \text{'question}, \text{'state}) \text{ state-pred where}$   
 $LSTP P \equiv \lambda s. P s\downarrow$

## 4.4 Control predicates

Following Lamport (1980)<sup>1</sup>, we define the *at* predicate, which holds of a process when control resides at that location. Due to non-determinism processes can be *at* a set of locations; it is more like “a statement with this location is enabled”, which incidentally handles non-unique locations. Lamport’s language is deterministic, so he doesn’t have this problem. This also allows him to develop a stronger theory about his control predicates.

**type-synonym** *'location label* = *'location set*

### primrec

$atC :: ('answer, 'location, 'question, 'state) com \Rightarrow 'location\ label$

### where

$atC (\{l\} Request\ action\ val) = \{l\}$   
 $| atC (\{l\} Response\ action) = \{l\}$   
 $| atC (\{l\} LocalOp\ f) = \{l\}$   
 $| atC (\{l\} IF - THEN - FI) = \{l\}$   
 $| atC (\{l\} IF - THEN - ELSE - FI) = \{l\}$   
 $| atC (\{l\} WHILE - DO - OD) = \{l\}$   
 $| atC (LOOP\ DO\ c\ OD) = atC\ c$   
 $| atC (c1;; c2) = atC\ c1$   
 $| atC (c1 \oplus c2) = atC\ c1 \cup atC\ c2$

**primrec**  $atCs :: ('answer, 'location, 'question, 'state) com\ list \Rightarrow 'location\ label\ \mathbf{where}$

$atCs [] = \{\}$   
 $| atCs (c \# -) = atC\ c$

We provide the following definitions to the end-user.

*AT* maps process names to a predicate that is true of locations where control for that process resides, and the abbreviation *at* provides a conventional way to use it. The constant *atS* specifies that control for process *p* resides at one of the given locations. This stands in for, and generalises, the *in* predicate of Lamport (1980).

**definition**  $AT :: ('answer, 'location, 'proc, 'question, 'state) system\ state \Rightarrow 'proc \Rightarrow 'location\ label\ \mathbf{where}$

$AT\ s\ p = atCs\ (cPGM\ (GST\ s\ p))$

**abbreviation**  $at :: 'proc \Rightarrow 'location \Rightarrow ('answer, 'location, 'proc, 'question, 'state) state\ pred\ \mathbf{where}$

$at\ p\ l\ s \equiv l \in AT\ s\ p$

**definition**  $atS :: 'proc \Rightarrow 'location\ set \Rightarrow ('answer, 'location, 'proc, 'question, 'state) state\ pred\ \mathbf{where}$

$atS\ p\ ls\ s = (\exists l \in ls. at\ p\ l\ s)$

**definition**  $atLs :: 'proc \Rightarrow 'location\ label\ set \Rightarrow ('answer, 'location, 'proc, 'question, 'state) state\ pred\ \mathbf{where}$

$atLs\ p\ labels\ s = (AT\ s\ p \in labels)$

**abbreviation** (input)  $atL :: 'proc \Rightarrow 'location\ label \Rightarrow ('answer, 'location, 'proc, 'question, 'state) state\ pred\ \mathbf{where}$

$atL\ p\ label \equiv atLs\ p\ \{label\}$

**definition**  $atPLs :: ('proc \times 'location\ label) set \Rightarrow ('answer, 'location, 'proc, 'question, 'state) state\ pred\ \mathbf{where}$

$atPLs\ pls = (\forall p\ label. \langle(p, label) \in pls \rangle \longrightarrow atL\ p\ label)$

The constant *taken* provides a way of identifying which transition was taken. It is somewhat like Lamport’s *after*, but not quite due to the presence of non-determinism here. This does not work well for invariants or preconditions.

**definition**  $taken :: 'proc \Rightarrow 'location \Rightarrow ('answer, 'location, 'proc, 'question, 'state) state\ pred\ \mathbf{where}$

$taken\ p\ l\ s \longleftrightarrow cTKN\ (GST\ s\ p) = Some\ l$

<sup>1</sup>Manna and Pnueli (1995) also develop a theory of locations. I think Lamport attributes control predicates to Owicki in her PhD thesis (under Gries). I did not find a treatment of procedures. Manna and Pnueli (1991) observe that a notation for making assertions over sets of locations reduces clutter significantly.

A process is terminated if it not at any control location.

**abbreviation**  $(input) \text{ terminated} :: 'proc \Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ state-pred}$  **where**  
 $\text{terminated } p \equiv \text{atL } p \ \{\}$

A complete system consists of one program per process, and a (global) constraint on their initial local states. From these we can construct the set of initial global states and all those reachable by system steps (§4.3).

**type-synonym**  $( 'answer, 'location, 'proc, 'question, 'state) \text{ programs}$   
 $= 'proc \Rightarrow ('answer, 'location, 'question, 'state) \text{ com}$

**record**  $( 'answer, 'location, 'proc, 'question, 'state) \text{ pre-system} =$   
 $\text{PGMs} :: ('answer, 'location, 'proc, 'question, 'state) \text{ programs}$   
 $\text{INIT} :: ('proc, 'state) \text{ local-state-pred}$

**definition**

$\text{initial-state} :: ('answer, 'location, 'proc, 'question, 'state, 'ext) \text{ pre-system-ext}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ global-state}$   
 $\Rightarrow \text{bool}$

**where**

$\text{initial-state } \text{sys } s = ((\forall p. \text{cPGM } (s \ p) = [\text{PGMs } \text{sys } p] \wedge \text{cTKN } (s \ p) = \text{None}) \wedge \text{INIT } \text{sys } (\text{cLST } \circ s))$

We construct infinite runs of a system by allowing stuttering, i.e., arbitrary repetitions of states following Lamport (2002, Chapter 8), by taking the reflexive closure of the *system-step* relation. Therefore terminated programs infinitely repeat their final state (but note our definition of terminated processes in §4.4).

Some accounts define stuttering as the *finite* repetition of states. With or without this constraint *prerun* contains *junk* in the form of unfair runs, where particular processes do not progress.

**definition**

$\text{system-step-reflclp} :: ('answer, 'location, 'proc, 'question, 'state) \text{ system-state seq-pred}$

**where**

$\text{system-step-reflclp } \sigma \longleftrightarrow (\lambda sh \ sh'. \exists pls. \text{system-step } pls \ sh' \ sh) \text{==} (\sigma \ 0) (\sigma \ 1)$

**definition**

$\text{prerun} :: ('answer, 'location, 'proc, 'question, 'state, 'ext) \text{ pre-system-ext}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ system-state seq-pred}$

**where**

$\text{prerun } \text{sys} = ((\lambda \sigma. \text{initial-state } \text{sys } (\text{GST } (\sigma \ 0))) \wedge \text{HST } (\sigma \ 0) = []) \wedge \Box \text{system-step-reflclp}$

**definition** — state-based invariants only

$\text{prerun-valid} :: ('answer, 'location, 'proc, 'question, 'state, 'ext) \text{ pre-system-ext}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ state-pred} \Rightarrow \text{bool} \ (- \models_{\text{pre}} - [11, 0] \ 11)$

**where**

$(\text{sys} \models_{\text{pre}} \varphi) \longleftrightarrow (\forall \sigma. \text{prerun } \text{sys } \sigma \longrightarrow (\Box [\varphi]) \ \sigma)$

A *run* of a system is a *prerun* that satisfies the *FAIR* requirement. Typically this would include *weak fairness* for every transition of every process.

**record**  $( 'answer, 'location, 'proc, 'question, 'state) \text{ system} =$   
 $( 'answer, 'location, 'proc, 'question, 'state) \text{ pre-system}$   
 $+ \text{FAIR} :: ('answer, 'location, 'proc, 'question, 'state) \text{ system-state seq-pred}$

**definition**

$\text{run} :: ('answer, 'location, 'proc, 'question, 'state) \text{ system}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ system-state seq-pred}$

**where**

$\text{run } \text{sys} = (\text{prerun } \text{sys} \wedge \text{FAIR } \text{sys})$

**definition**

$\text{valid} :: ('answer, 'location, 'proc, 'question, 'state) \text{ system}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ system-state seq-pred} \Rightarrow \text{bool} \ (- \models - [11, 0] \ 11)$

where

$$(sys \models \varphi) \longleftrightarrow (\forall \sigma. \text{run sys } \sigma \longrightarrow \varphi \sigma)$$

## 5 State-based invariants

We provide a simple-minded verification condition generator (VCG) for this language, providing support for establishing state-based invariants. It is just one way of reasoning about CIMP programs and is proven sound wrt to the CIMP semantics.

Our approach follows Lamport (1980); Lamport and Schneider (1984) (and the later Lamport (2002)) and closely related work by Apt, Francez, and de Roever (1980), Cousot and Cousot (1980) and Levin and Gries (1981), who suggest the incorporation of a history variable. Cousot and Cousot (1980) apparently contains a completeness proof. Lamport mentions that this technique was well-known in the mid-80s when he proposed the use of prophecy variables<sup>2</sup>. See also de Roever, de Boer, Hannemann, Hooman, Lakhnech, Poel, and Zwiers (2001) for an extended discussion of some of this.

**declare** *small-step.intros*[*intro*]

**inductive-cases** *small-step-inv*:

$$\begin{aligned} & \{\!|l|\!\} \text{ Request action val } \# \text{ cs, ls } \rightarrow_a s' \\ & \{\!|l|\!\} \text{ Response action } \# \text{ cs, ls } \rightarrow_a s' \\ & \{\!|l|\!\} \text{ LocalOp } R \# \text{ cs, ls } \rightarrow_a s' \\ & \{\!|l|\!\} \text{ IF } b \text{ THEN } c \text{ FI } \# \text{ cs, ls } \rightarrow_a s' \\ & \{\!|l|\!\} \text{ IF } b \text{ THEN } c1 \text{ ELSE } c2 \text{ FI } \# \text{ cs, ls } \rightarrow_a s' \\ & \{\!|l|\!\} \text{ WHILE } b \text{ DO } c \text{ OD } \# \text{ cs, ls } \rightarrow_a s' \\ & \{\!|l|\!\} \text{ LOOP DO } c \text{ OD } \# \text{ cs, ls } \rightarrow_a s' \end{aligned}$$

**lemma** *small-step-stuck*:

$$\neg (\llbracket \cdot \rrbracket, s) \rightarrow_\alpha c'$$

**by** (*auto elim: small-step.cases*)

**declare** *system-step.intros*[*intro*]

By default we ask the simplifier to rewrite *atS* using ambient *AT* information.

**lemma** *atS-state-weak-cong*[*cong*]:

$$AT \ s \ p = AT \ s' \ p \implies atS \ p \ ls \ s \longleftrightarrow atS \ p \ ls \ s'$$

**by** (*auto simp: atS-def*)

We provide an incomplete set of basic rules for label sets.

**lemma** *atS-simps*:

$$\begin{aligned} & \neg atS \ p \ \{\} \ s \\ & atS \ p \ \{l\} \ s \longleftrightarrow at \ p \ l \ s \\ & \llbracket at \ p \ l \ s; l \in ls \rrbracket \implies atS \ p \ ls \ s \\ & (\forall l. at \ p \ l \ s \longrightarrow l \notin ls) \implies \neg atS \ p \ ls \ s \end{aligned}$$

**by** (*auto simp: atS-def*)

**lemma** *atS-mono*:

$$\llbracket atS \ p \ ls \ s; ls \subseteq ls' \rrbracket \implies atS \ p \ ls' \ s$$

**by** (*auto simp: atS-def*)

**lemma** *atS-un*:

$$atS \ p \ (l \cup l') \ s \longleftrightarrow atS \ p \ l \ s \vee atS \ p \ l' \ s$$

**by** (*auto simp: atS-def*)

**lemma** *atLs-disj-union*[*simp*]:

$$(atLs \ p \ label0 \vee atLs \ p \ label1) = atLs \ p \ (label0 \cup label1)$$

**unfolding** *atLs-def* **by** *simp*

---

<sup>2</sup><https://lamport.azurewebsites.net/pubs/pubs.html>

**lemma** *atLs-insert-disj*:

$atLs\ p\ (insert\ l\ label0) = (atL\ p\ l \vee atLs\ p\ label0)$

**by** *simp*

**lemma** *small-step-terminated*:

$s \rightarrow_x s' \implies atCs\ (fst\ s) = \{\} \implies atCs\ (fst\ s') = \{\}$

**by** (*induct pred: small-step*) *auto*

**lemma** *atC-not-empty*:

$atC\ c \neq \{\}$

**by** (*induct c*) *auto*

**lemma** *atCs-empty*:

$atCs\ cs = \{\} \longleftrightarrow cs = []$

**by** (*induct cs*) (*auto simp: atC-not-empty*)

**lemma** *terminated-no-commands*:

**assumes** *terminated p sh*

**shows**  $\exists s. GST\ sh\ p = ([], s)$

**using** *assms unfolding atLs-def AT-def* **by** (*metis atCs-empty prod.collapse singletonD*)

**lemma** *terminated-GST-stable*:

**assumes** *system-step q sh' sh*

**assumes** *terminated p sh*

**shows**  $GST\ sh\ p = GST\ sh'\ p$

**using** *assms* **by** (*auto dest!: terminated-no-commands simp: small-step-stuck elim!: system-step.cases*)

**lemma** *terminated-stable*:

**assumes** *system-step q sh' sh*

**assumes** *terminated p sh*

**shows** *terminated p sh'*

**using** *assms unfolding atLs-def AT-def*

**by** (*fastforce split: if-splits prod.splits*

*dest: small-step-terminated*

*elim!: system-step.cases*)

**lemma** *system-step-pls-nonempty*:

**assumes** *system-step pls sh' sh*

**shows**  $pls \neq \{\}$

**using** *assms* **by** *cases simp-all*

**lemma** *system-step-no-change*:

**assumes** *system-step ps sh' sh*

**assumes**  $p \notin ps$

**shows**  $GST\ sh'\ p = GST\ sh\ p$

**using** *assms* **by** *cases simp-all*

**lemma** *initial-stateD*:

**assumes** *initial-state sys s*

**shows**  $AT\ (\Downarrow GST = s, HST = []) = atC \circ PGMs\ sys \wedge INIT\ sys\ (\Downarrow GST = s, HST = []) \downarrow \wedge (\forall p\ l. \neg taken\ p\ l\ (\Downarrow GST = s, HST = []))$

**using** *assms unfolding initial-state-def split-def o-def LST-def AT-def taken-def* **by** *simp*

**lemma** *initial-states-initial[iff]*:

**assumes** *initial-state sys s*

**shows**  $at\ p\ l\ (\Downarrow GST = s, HST = []) \longleftrightarrow l \in atC\ (PGMs\ sys\ p)$

**using** *assms unfolding initial-state-def split-def AT-def* **by** *simp*

**definition**

*reachable-state* :: ('answer, 'location, 'proc, 'question, 'state, 'ext) pre-system-ext  
 $\Rightarrow$  ('answer, 'location, 'proc, 'question, 'state) state-pred

**where**

*reachable-state sys s*  $\longleftrightarrow$  ( $\exists \sigma i$ . prerun sys  $\sigma \wedge \sigma i = s$ )

**lemma** *reachable-stateE*:

**assumes** *reachable-state sys sh*  
**assumes**  $\bigwedge \sigma i$ . prerun sys  $\sigma \implies P (\sigma i)$   
**shows** *P sh*

**using** *assms unfolding reachable-state-def by blast*

**lemma** *prerun-reachable-state*:

**assumes** *prerun sys  $\sigma$*   
**shows** *reachable-state sys ( $\sigma i$ )*

**using** *assms unfolding prerun-def LTL.defs system-step-reflclp-def reachable-state-def by auto*

**lemma** *reachable-state-induct*[*consumes 1, case-names init LocalStep CommunicationStep, induct set: reachable-state*]:

**assumes** *r: reachable-state sys sh*  
**assumes** *i:  $\bigwedge s$ . initial-state sys s  $\implies P (\langle GST = s, HST = [] \rangle)$*   
**assumes** *l:  $\bigwedge sh ls' p$ .  $\llbracket$ reachable-state sys sh; P sh; GST sh p  $\rightarrow_{\tau}$  ls' $\rrbracket \implies P (\langle GST = (GST sh)(p := ls'), HST = HST sh \rangle)$*   
**assumes** *c:  $\bigwedge sh ls1' ls2' p1 p2 \alpha \beta$ .  
 $\llbracket$ reachable-state sys sh; P sh;  
GST sh p1  $\rightarrow_{\langle \alpha, \beta \rangle} ls1'$ ; GST sh p2  $\rightarrow_{\langle \alpha, \beta \rangle} ls2'$ ; p1  $\neq$  p2  $\rrbracket$   
 $\implies P (\langle GST = (GST sh)(p1 := ls1', p2 := ls2'), HST = HST sh @ [(\alpha, \beta)] \rangle)$*

**shows** *P sh*

**using** *r*

**proof**(*rule reachable-stateE*)

**fix**  $\sigma i$  **assume** *prerun sys  $\sigma$*  **show** *P ( $\sigma i$ )*

**proof**(*induct i*)

**case** 0 **from**  $\langle$ prerun sys  $\sigma$  $\rangle$  **show** ?*case*

**unfolding** *prerun-def by (metis (full-types) i old.unit.exhaust system-state.surjective)*

**next**

**case** (*Suc i*) **with**  $\langle$ prerun sys  $\sigma$  $\rangle$  **show** ?*case*

**unfolding** *prerun-def LTL.defs system-step-reflclp-def reachable-state-def*

**apply** *clarsimp*

**apply** (*drule-tac x=i in spec*)

**apply** (*erule disjE; clarsimp*)

**apply** (*erule system-step.cases; clarsimp*)

**apply** (*metis (full-types)  $\langle$ prerun sys  $\sigma$  $\rangle$  l old.unit.exhaust prerun-reachable-state system-state.surjective*)

**apply** (*metis (full-types)  $\langle$ prerun sys  $\sigma$  $\rangle$  c old.unit.exhaust prerun-reachable-state system-state.surjective*)

**done**

**qed**

**qed**

**lemma** *prerun-valid-TrueI*:

**shows** *sys  $\models_{pre}$   $\langle$ True $\rangle$*

**unfolding** *prerun-valid-def by simp*

**lemma** *prerun-valid-conjI*:

**assumes** *sys  $\models_{pre}$  P*

**assumes** *sys  $\models_{pre}$  Q*

**shows** *sys  $\models_{pre}$  P  $\wedge$  Q*

**using** *assms unfolding prerun-valid-def always-def by simp*

**lemma** *valid-prerun-lift*:

**assumes**  $sys \models_{pre} I$

**shows**  $sys \models \Box[I]$

**using** *assms unfolding prerun-valid-def valid-def run-def* **by** *blast*

**lemma** *prerun-valid-induct*:

**assumes**  $\bigwedge \sigma. \text{prerun } sys \ \sigma \implies [I] \ \sigma$

**assumes**  $\bigwedge \sigma. \text{prerun } sys \ \sigma \implies ([I] \leftrightarrow (\bigcirc[I])) \ \sigma$

**shows**  $sys \models_{pre} I$

**unfolding** *prerun-valid-def* **using** *assms* **by** (*simp add: always-induct*)

**lemma** *prerun-validI*:

**assumes**  $\bigwedge s. \text{reachable-state } sys \ s \implies I \ s$

**shows**  $sys \models_{pre} I$

**unfolding** *prerun-valid-def* **using** *assms* **by** (*simp add: alwaysI prerun-reachable-state*)

**lemma** *prerun-validE*:

**assumes** *reachable-state*  $sys \ s$

**assumes**  $sys \models_{pre} I$

**shows**  $I \ s$

**using** *assms unfolding prerun-valid-def*

**by** (*metis alwaysE reachable-stateE suffix-state-prop*)

### 5.0.1 Relating reachable states to the initial programs

To usefully reason about the control locations presumably embedded in the single global invariant, we need to link the programs we have in reachable state  $s$  to the programs in the initial states. The *fragments* function decomposes the program into statements that can be directly executed (§??). We also compute the locations we could be at after executing that statement as a function of the process's local state.

Eliding the bodies of *IF* and *WHILE* statements yields smaller (but equivalent) proof obligations.

**type-synonym** (*'answer, 'location, 'question, 'state*) *loc-comp*

= *'state*  $\Rightarrow$  *'location set*

**fun** *lconst* :: *'location set*  $\Rightarrow$  (*'answer, 'location, 'question, 'state*) *loc-comp* **where**

*lconst*  $lp \ s = lp$

**definition** *lcond* :: *'location set*  $\Rightarrow$  *'location set*  $\Rightarrow$  *'state* *bxp*

$\Rightarrow$  (*'answer, 'location, 'question, 'state*) *loc-comp* **where**

*lcond*  $lp \ lp' \ b \ s = (\text{if } b \ s \ \text{then } lp \ \text{else } lp')$

**lemma** *lcond-split*:

$Q \ (lcond \ lp \ lp' \ b \ s) \longleftrightarrow (b \ s \longrightarrow Q \ lp) \wedge (\neg b \ s \longrightarrow Q \ lp')$

**unfolding** *lcond-def* **by** (*simp split: if-splits*)

**lemma** *lcond-split-asm*:

$Q \ (lcond \ lp \ lp' \ b \ s) \longleftrightarrow \neg ((b \ s \wedge \neg Q \ lp) \vee (\neg b \ s \wedge \neg Q \ lp'))$

**unfolding** *lcond-def* **by** (*simp split: if-splits*)

**lemmas** *lcond-splits = lcond-split lcond-split-asm*

**fun**

*fragments* :: (*'answer, 'location, 'question, 'state*) *com*

$\Rightarrow$  *'location set*

$\Rightarrow$  ( (*'answer, 'location, 'question, 'state*) *com*

$\times$  (*'answer, 'location, 'question, 'state*) *loc-comp* ) *set*

**where**

*fragments* ( $\{\!|l|\!\}$  *IF*  $b$  *THEN*  $c$  *FI*) *aft*

$$\begin{aligned}
&= \{ (\{l\} \text{ IF } b \text{ THEN } c' \text{ FI, } l\text{cond } (atC \ c) \text{ aft } b) \mid c'. \text{ True } \} \\
&\cup \text{ fragments } c \text{ aft} \\
| \text{ fragments } (\{l\} \text{ IF } b \text{ THEN } c1 \text{ ELSE } c2 \text{ FI}) \text{ aft} \\
&= \{ (\{l\} \text{ IF } b \text{ THEN } c1' \text{ ELSE } c2' \text{ FI, } l\text{cond } (atC \ c1) \text{ (atC } c2) \ b) \mid c1' \ c2'. \text{ True } \} \\
&\cup \text{ fragments } c1 \text{ aft} \cup \text{ fragments } c2 \text{ aft} \\
| \text{ fragments } (\text{ LOOP DO } c \text{ OD}) \text{ aft} &= \text{ fragments } c \text{ (atC } c) \\
| \text{ fragments } (\{l\} \text{ WHILE } b \text{ DO } c \text{ OD}) \text{ aft} \\
&= \text{ fragments } c \ \{l\} \cup \{ (\{l\} \text{ WHILE } b \text{ DO } c' \text{ OD, } l\text{cond } (atC \ c) \text{ aft } b) \mid c'. \text{ True } \} \\
| \text{ fragments } (c1;; c2) \text{ aft} &= \text{ fragments } c1 \text{ (atC } c2) \cup \text{ fragments } c2 \text{ aft} \\
| \text{ fragments } (c1 \oplus c2) \text{ aft} &= \text{ fragments } c1 \text{ aft} \cup \text{ fragments } c2 \text{ aft} \\
| \text{ fragments } c \text{ aft} &= \{ (c, l\text{const aft}) \}
\end{aligned}$$

### fun

*fragmentsL* :: ('answer, 'location, 'question, 'state) com list  
 $\Rightarrow$  ( ('answer, 'location, 'question, 'state) com  
 $\times$  ('answer, 'location, 'question, 'state) loc-comp ) set

### where

*fragmentsL* [] = {}  
| *fragmentsL* [c] = *fragments* c {}  
| *fragmentsL* (c # c' # cs) = *fragments* c (atC c')  $\cup$  *fragmentsL* (c' # cs)

### abbreviation

*fragmentsLS* :: ('answer, 'location, 'question, 'state) local-state  
 $\Rightarrow$  ( ('answer, 'location, 'question, 'state) com  
 $\times$  ('answer, 'location, 'question, 'state) loc-comp ) set

### where

*fragmentsLS* s  $\equiv$  *fragmentsL* (cPGM s)

We show that taking system steps preserves fragments.

### lemma small-step-fragmentsLS:

**assumes**  $s \rightarrow_{\alpha} s'$   
**shows** *fragmentsLS* s'  $\subseteq$  *fragmentsLS* s  
**using** *assms* **by** *induct* (case-tac [!] cs, auto)

### lemma reachable-state-fragmentsLS:

**assumes** *reachable-state* sys sh  
**shows** *fragmentsLS* (GST sh p)  $\subseteq$  *fragments* (PGMs sys p) {}  
**using** *assms*  
**by** (*induct* rule: *reachable-state-induct*)  
(auto simp: *initial-state-def* dest: *subsetD*[OF *small-step-fragmentsLS*])

### inductive

*basic-com* :: ('answer, 'location, 'question, 'state) com  $\Rightarrow$  bool

### where

*basic-com* (\{l\} Request action val)  
| *basic-com* (\{l\} Response action)  
| *basic-com* (\{l\} LocalOp R)  
| *basic-com* (\{l\} IF b THEN c FI)  
| *basic-com* (\{l\} IF b THEN c1 ELSE c2 FI)  
| *basic-com* (\{l\} WHILE b DO c OD)

### lemma fragments-basic-com:

**assumes** (c', aft')  $\in$  *fragments* c aft  
**shows** *basic-com* c'  
**using** *assms* **by** (*induct* c arbitrary: aft) (auto intro: *basic-com.intros*)

### lemma fragmentsL-basic-com:

**assumes** (c', aft')  $\in$  *fragmentsL* cs



**shows** *basic-com c'*  
**using** *assms*  
**apply** (*induct cs*)  
**apply** *simp*  
**apply** (*case-tac cs*)  
**apply** (*auto simp: fragments-basic-com*)  
**done**

To reason about system transitions we need to identify which basic statement gets executed next. To that end we factor out the recursive cases of the *small-step* semantics into *contexts*, which isolate the *basic-com* commands with immediate externally-visible behaviour. Note that non-determinism means that more than one *basic-com* can be enabled at a time.

The representation of evaluation contexts follows Berghofer (2012). This style of operational semantics was originated by Felleisen and Hieb (1992).

**type-synonym** (*'answer, 'location, 'question, 'state*) *ctxt*  
 $= ((\text{'answer, 'location, 'question, 'state}) \text{com} \Rightarrow (\text{'answer, 'location, 'question, 'state}) \text{com})$   
 $\times ((\text{'answer, 'location, 'question, 'state}) \text{com} \Rightarrow (\text{'answer, 'location, 'question, 'state}) \text{com list})$

### inductive-set

*ctxt* :: (*'answer, 'location, 'question, 'state*) *ctxt* set

#### where

*C-Hole*: (*id, <[]>*)  $\in$  *ctxt*

| *C-Loop*: (*E, fctx*)  $\in$  *ctxt*  $\implies (\lambda c1. \text{LOOP DO } E \ c1 \ OD, \lambda c1. \text{fctx } c1 \ @ \ [\text{LOOP DO } E \ c1 \ OD]) \in \text{ctxt}$

| *C-Seq*: (*E, fctx*)  $\in$  *ctxt*  $\implies (\lambda c1. E \ c1;; \ c2, \lambda c1. \text{fctx } c1 \ @ \ [c2]) \in \text{ctxt}$

| *C-Choose1*: (*E, fctx*)  $\in$  *ctxt*  $\implies (\lambda c1. E \ c1 \oplus \ c2, \text{fctx}) \in \text{ctxt}$

| *C-Choose2*: (*E, fctx*)  $\in$  *ctxt*  $\implies (\lambda c2. c1 \oplus \ E \ c2, \text{fctx}) \in \text{ctxt}$

We can decompose a small step into a context and a *basic-com*.

### fun

*decompose-com* :: (*'answer, 'location, 'question, 'state*) *com*  
 $\Rightarrow ((\text{'answer, 'location, 'question, 'state}) \text{com}$   
 $\times (\text{'answer, 'location, 'question, 'state}) \text{ctxt}) \text{set}$

#### where

*decompose-com* (*LOOP DO c1 OD*) = { (*c, λt. LOOP DO ictxt t OD, λt. fctx t @ [LOOP DO ictxt t OD]*) | *c fctx ictxt. (c, ictxt, fctx) ∈ decompose-com c1* }

| *decompose-com* (*c1;; c2*) = { (*c, λt. ictxt t;; c2, λt. fctx t @ [c2]*) | *c fctx ictxt. (c, ictxt, fctx) ∈ decompose-com c1* }

| *decompose-com* (*c1 ⊕ c2*) = { (*c, λt. ictxt t ⊕ c2, fctx*) | *c fctx ictxt. (c, ictxt, fctx) ∈ decompose-com c1* }  
 $\cup \{ (c, \lambda t. c1 \oplus \text{ictxt } t, \text{fctx}) \mid c \text{ fctx } \text{ictxt}. (c, \text{ictxt}, \text{fctx}) \in \text{decompose-com } c2 \}$

| *decompose-com* *c* = {(*c, id, <[]>*)}

### definition

*decomposeLS* :: (*'answer, 'location, 'question, 'state*) *local-state*  
 $\Rightarrow ((\text{'answer, 'location, 'question, 'state}) \text{com}$   
 $\times ((\text{'answer, 'location, 'question, 'state}) \text{com} \Rightarrow (\text{'answer, 'location, 'question, 'state}) \text{com})$   
 $\times ((\text{'answer, 'location, 'question, 'state}) \text{com} \Rightarrow (\text{'answer, 'location, 'question, 'state}) \text{com list}) \text{set}$

#### where

*decomposeLS* *s* = (*case cPGM s of c # - => decompose-com c | - => {}*)

### lemma *ctxt-inj*:

**assumes** (*E, fctx*)  $\in$  *ctxt*

**assumes** *E x = E y*

**shows** *x = y*

**using** *assms* **by** (*induct set: ctxt*) *auto*

**lemma** *decompose-com-non-empty*: *decompose-com c*  $\neq \{\}$

**by** (*induct c*) *auto*

**lemma** *decompose-com-basic-com*:

**assumes**  $(c', \text{ctxts}) \in \text{decompose-com } c$

**shows** *basic-com*  $c'$

**using** *assms* **by** (*induct*  $c$  *arbitrary*:  $c'$  *ctxts*) (*auto intro*: *basic-com.intros*)

**lemma** *decomposeLS-basic-com*:

**assumes**  $(c', \text{ctxts}) \in \text{decomposeLS } s$

**shows** *basic-com*  $c'$

**using** *assms* **unfolding** *decomposeLS-def* **by** (*simp add*: *decompose-com-basic-com split*: *list.splits*)

**lemma** *decompose-com-ctxt*:

**assumes**  $(c', \text{ctxts}) \in \text{decompose-com } c$

**shows**  $\text{ctxts} \in \text{ctxt}$

**using** *assms* **by** (*induct*  $c$  *arbitrary*:  $c'$  *ctxts*) (*auto intro*: *ctxt.intros*)

**lemma** *decompose-com-ictxt*:

**assumes**  $(c', \text{ictxt}, \text{fctxt}) \in \text{decompose-com } c$

**shows**  $\text{ictxt } c' = c$

**using** *assms* **by** (*induct*  $c$  *arbitrary*:  $c'$  *ictxt* *fctxt*) *auto*

**lemma** *decompose-com-small-step*:

**assumes** *as*:  $(c' \# \text{fctxt } c' @ \text{cs}, s) \rightarrow_{\alpha} s'$

**assumes** *ds*:  $(c', \text{ictxt}, \text{fctxt}) \in \text{decompose-com } c$

**shows**  $(c \# \text{cs}, s) \rightarrow_{\alpha} s'$

**using** *decompose-com-ctxt*[*OF ds*] *as* *decompose-com-ictxt*[*OF ds*]

**by** (*induct* *ictxt* *fctxt* *arbitrary*:  $c$  *cs*)

(*cases*  $s'$ , *fastforce* *simp*: *fun-eq-iff* *dest*: *ctxt-inj*)**+**

**theorem** *context-decompose*:

$s \rightarrow_{\alpha} s' \iff (\exists (c, \text{ictxt}, \text{fctxt}) \in \text{decomposeLS } s.$

$c\text{PGM } s = \text{ictxt } c \# \text{tl } (c\text{PGM } s)$

$\wedge (c \# \text{fctxt } c @ \text{tl } (c\text{PGM } s), c\text{TKN } s, c\text{LST } s) \rightarrow_{\alpha} s'$

$\wedge (\forall l \in \text{at } C \ c. c\text{TKN } s' = \text{Some } l))$  (**is** *?lhs* = *?rhs*)

**proof**(*rule iffI*)

**assume** *?lhs* **then show** *?rhs*

**unfolding** *decomposeLS-def*

**proof**(*induct* *rule*: *small-step.induct*)

**case** (*Choose1*  $c1$   $cs$   $s \alpha$   $cs' s' c2$ ) **then show** *?case*

**apply** *clarsimp*

**apply** (*rename-tac*  $c$  *ictxt* *fctxt*)

**apply** (*rule-tac*  $x=(c, \lambda t. \text{ictxt } t \oplus c2, \text{fctxt})$  **in** *beXI*)

**apply** *auto*

**done**

**next**

**case** (*Choose2*  $c2$   $cs$   $s \alpha$   $cs' s' c1$ ) **then show** *?case*

**apply** *clarsimp*

**apply** (*rename-tac*  $c$  *ictxt* *fctxt*)

**apply** (*rule-tac*  $x=(c, \lambda t. c1 \oplus \text{ictxt } t, \text{fctxt})$  **in** *beXI*)

**apply** *auto*

**done**

**qed** *fastforce***+**

**next**

**assume** *?rhs* **then show** *?lhs*

**unfolding** *decomposeLS-def*

**by** (*cases*  $s$ ) (*auto split*: *list.splits* *dest*: *decompose-com-small-step*)

**qed**

While we only use this result left-to-right (to decompose a small step into a basic one), this equivalence shows

that we lose no information in doing so.

Decomposing a compound command preserves *fragments* too.

**fun**

```
loc-compC :: ('answer, 'location, 'question, 'state) com
           ⇒ ('answer, 'location, 'question, 'state) com list
           ⇒ ('answer, 'location, 'question, 'state) loc-comp
```

**where**

```
loc-compC (⟦l⟧ IF b THEN c FI) cs = lcond (atC c) (atCs cs) b
| loc-compC (⟦l⟧ IF b THEN c1 ELSE c2 FI) cs = lcond (atC c1) (atC c2) b
| loc-compC (LOOP DO c OD) cs = lconst (atC c)
| loc-compC (⟦l⟧ WHILE b DO c OD) cs = lcond (atC c) (atCs cs) b
| loc-compC c cs = lconst (atCs cs)
```

**lemma** *decompose-fragments*:

```
assumes (c, ictxt, fctxt) ∈ decompose-com c0
shows (c, loc-compC c (fctxt c @ cs)) ∈ fragments c0 (atCs cs)
```

**using** *assms*

**proof**(*induct* c0 *arbitrary*: c ictxt fctxt cs)

```
case (Loop c01 c ictxt fctxt cs)
```

```
from Loop.premis Loop.hyps(1)[where cs=ictxt c # cs] show ?case by (auto simp: decompose-com-ictxt)
```

**next**

```
case (Seq c01 c02 c ictxt fctxt cs)
```

```
from Seq.premis Seq.hyps(1)[where cs=c02 # cs] show ?case by auto
```

**qed** *auto*

**lemma** *at-decompose*:

```
assumes (c, ictxt, fctxt) ∈ decompose-com c0
shows atC c ⊆ atC c0
```

**using** *assms* **by** (*induct* c0 *arbitrary*: c ictxt fctxt; *fastforce*)

**lemma** *at-decomposeLS*:

```
assumes (c, ictxt, fctxt) ∈ decomposeLS s
shows atC c ⊆ atCs (cPGM s)
```

**using** *assms* **unfolding** *decomposeLS-def* **by** (*auto simp*: *at-decompose split*: *list.splits*)

**lemma** *decomposeLS-fragmentsLS*:

```
assumes (c, ictxt, fctxt) ∈ decomposeLS s
shows (c, loc-compC c (fctxt c @ tl (cPGM s))) ∈ fragmentsLS s
```

**using** *assms*

**proof**(*cases* cPGM s)

```
case (Cons d ds)
```

```
with assms decompose-fragments[where cs=ds] show ?thesis
```

```
by (cases ds) (auto simp: decomposeLS-def)
```

**qed** (*simp add*: *decomposeLS-def*)

**lemma** *small-step-loc-compC*:

```
assumes basic-com c
```

```
assumes (c # cs, ls) →α ls'
```

```
shows loc-compC c cs (snd ls) = atCs (cPGM ls')
```

**using** *assms* **by** (*fastforce elim*: *basic-com.cases elim*!: *small-step-inv split*: *lcond-splits*)

The headline result allows us to constrain the initial and final states of a given small step in terms of the original programs, provided the initial state is reachable.

**theorem** *decompose-small-step*:

```
assumes GST sh p →α ps'
```

```
assumes reachable-state sys sh
```

```
obtains c cs aft
```

**where**  $(c, aft) \in \text{fragments}(\text{PGMs } sys \ p) \ \{\}$   
**and**  $atC \ c \subseteq atCs \ (cPGM \ (GST \ sh \ p))$   
**and**  $aft \ (cLST \ (GST \ sh \ p)) = atCs \ (cPGM \ ps')$   
**and**  $(c \ \# \ cs, cTKN \ (GST \ sh \ p), cLST \ (GST \ sh \ p)) \rightarrow_\alpha \ ps'$   
**and**  $\forall l \in atC \ c. \ cTKN \ ps' = \text{Some } l$   
**using** *assms*  
**apply**  $-$   
**apply**  $(\text{frule } iffD1[\text{OF } \text{context-decompose}])$   
**apply** *clarsimp*  
**apply**  $(\text{frule } \text{decomposeLS-fragmentsLS})$   
**apply**  $(\text{frule } \text{at-decomposeLS})$   
**apply**  $(\text{frule } (1) \ \text{subsetD}[\text{OF } \text{reachable-state-fragmentsLS}])$   
**apply**  $(\text{frule } \text{decomposeLS-basic-com})$   
**apply**  $(\text{frule } (1) \ \text{small-step-loc-compC})$   
**apply** *simp*  
**done**

Reasoning by induction over the reachable states with *decompose-small-step* is quite tedious. We provide a very simple VCG that generates friendlier local proof obligations in §5.1.

## 5.1 Simple-minded Hoare Logic/VCG for CIMP

We do not develop a proper Hoare logic or full VCG for CIMP: this machinery merely packages up the subgoals that arise from induction over the reachable states (§5). This is somewhat in the spirit of [Ridge \(2009\)](#).

Note that this approach is not compositional: it consults the original system to find matching communicating pairs, and *aft* tracks the labels of possible successor statements. More serious Hoare logics are provided by [Cousot and Cousot \(1989\)](#); [Lamport \(1980\)](#); [Lamport and Schneider \(1984\)](#).

Intuitively we need to discharge a proof obligation for either *Requests* or *Responses* but not both. Here we choose to focus on *Requests* as we expect to have more local information available about these.

### inductive

$vcg :: ('answer, 'location, 'proc, 'question, 'state) \text{ programs}$   
 $\Rightarrow 'proc$   
 $\Rightarrow ('answer, 'location, 'question, 'state) \text{ loc-comp}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ state-pred}$   
 $\Rightarrow ('answer, 'location, 'question, 'state) \text{ com}$   
 $\Rightarrow ('answer, 'location, 'proc, 'question, 'state) \text{ state-pred}$   
 $\Rightarrow \text{bool } (-, -, - \vdash / \{-\} / - / \{-\}) [11,0,0,0,0,0] \ 11$

### where

$\llbracket \bigwedge aft' \ action' \ s \ ps' \ p's' \ l' \ \beta \ s' \ p'. \$   
 $\llbracket \text{pre } s; (\{l'\} \ \text{Response } \ action', \ aft') \in \text{fragments}(\text{coms } p') \ \{\}; \ p \neq p'; \$   
 $\text{ps}' \in \text{val } \beta \ (s \downarrow p); \ (p's', \ \beta) \in \text{action}' \ (\text{action} \ (s \downarrow p)) \ (s \downarrow p'); \$   
 $\text{at } p \ l \ s; \ \text{at } p' \ l' \ s; \$   
 $AT \ s' = (AT \ s)(p := aft \ (s \downarrow p), \ p' := aft' \ (s \downarrow p)); \$   
 $s' \downarrow = s \downarrow (p := ps', \ p' := p's'); \$   
 $\text{taken } p \ l \ s'; \$   
 $HST \ s' = HST \ s \ @ \ [(action \ (s \downarrow p), \ \beta)]; \$   
 $\forall p'' \in -\{p, p'\}. \ GST \ s' \ p'' = GST \ s \ p'' \$   
 $\rrbracket \implies \text{post } s' \$   
 $\rrbracket \implies \text{coms}, p, aft \vdash \{-pre\} \ \{l\} \ \text{Request } \ action \ \text{val } \ \{post\} \$   
 $\mid \llbracket \bigwedge s \ ps' \ s'. \$   
 $\llbracket \text{pre } s; \ ps' \in f \ (s \downarrow p); \$   
 $\text{at } p \ l \ s; \$   
 $AT \ s' = (AT \ s)(p := aft \ (s \downarrow p)); \$   
 $s' \downarrow = s \downarrow (p := ps'); \$   
 $\text{taken } p \ l \ s'; \$   
 $HST \ s' = HST \ s; \$   
 $\forall p'' \in -\{p\}. \ GST \ s' \ p'' = GST \ s \ p'' \$

$$\begin{array}{l}
\] \Longrightarrow post\ s' \\
\] \Longrightarrow coms, p, aft \vdash \{\{pre\}\ \{l\}\ LocalOp\ f\ \{post\}\} \\
| \llbracket \bigwedge s\ s'. \\
\llbracket pre\ s; \\
\quad at\ p\ l\ s; \\
\quad AT\ s' = (AT\ s)(p := aft\ (s \downarrow p)); \\
\quad s' \downarrow = s \downarrow; \\
\quad taken\ p\ l\ s'; \\
\quad HST\ s' = HST\ s; \\
\quad \forall p'' \in -\{p\}. GST\ s'\ p'' = GST\ s\ p'' \\
\] \Longrightarrow post\ s' \\
\] \Longrightarrow coms, p, aft \vdash \{\{pre\}\ \{l\}\ IF\ b\ THEN\ t\ FI\ \{post\}\} \\
| \llbracket \bigwedge s\ s'. \\
\llbracket pre\ s; \\
\quad at\ p\ l\ s; \\
\quad AT\ s' = (AT\ s)(p := aft\ (s \downarrow p)); \\
\quad s' \downarrow = s \downarrow; \\
\quad taken\ p\ l\ s'; \\
\quad HST\ s' = HST\ s; \\
\quad \forall p'' \in -\{p\}. GST\ s'\ p'' = GST\ s\ p'' \\
\] \Longrightarrow post\ s' \\
\] \Longrightarrow coms, p, aft \vdash \{\{pre\}\ \{l\}\ IF\ b\ THEN\ t\ ELSE\ e\ FI\ \{post\}\} \\
| \llbracket \bigwedge s\ s'. \\
\llbracket pre\ s; \\
\quad at\ p\ l\ s; \\
\quad AT\ s' = (AT\ s)(p := aft\ (s \downarrow p)); \\
\quad s' \downarrow = s \downarrow; \\
\quad taken\ p\ l\ s'; \\
\quad HST\ s' = HST\ s; \\
\quad \forall p'' \in -\{p\}. GST\ s'\ p'' = GST\ s\ p'' \\
\] \Longrightarrow post\ s' \\
\] \Longrightarrow coms, p, aft \vdash \{\{pre\}\ \{l\}\ WHILE\ b\ DO\ c\ OD\ \{post\}\}
\end{array}$$

— There are no proof obligations for the following commands, but including them makes some basic rules hold (§5.1.1):

$$\begin{array}{l}
| coms, p, aft \vdash \{\{pre\}\ \{l\}\ Response\ action\ \{post\}\} \\
| coms, p, aft \vdash \{\{pre\}\ c1\ ;;\ c2\ \{post\}\} \\
| coms, p, aft \vdash \{\{pre\}\ LOOP\ DO\ c\ OD\ \{post\}\} \\
| coms, p, aft \vdash \{\{pre\}\ c1\ \oplus\ c2\ \{post\}\}
\end{array}$$

We abbreviate invariance with one-sided validity syntax.

**abbreviation** *valid-inv*  $(-, -, - \vdash / \{-\} / - [11,0,0,0,0] 11)$  **where**  
 $coms, p, aft \vdash \{I\}\ c \equiv coms, p, aft \vdash \{I\}\ c\ \{I\}$

**inductive-cases** *vcg-inv*:

$$\begin{array}{l}
coms, p, aft \vdash \{\{pre\}\ \{l\}\ Request\ action\ val\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ \{l\}\ LocalOp\ f\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ \{l\}\ IF\ b\ THEN\ t\ FI\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ \{l\}\ IF\ b\ THEN\ t\ ELSE\ e\ FI\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ \{l\}\ WHILE\ b\ DO\ c\ OD\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ LOOP\ DO\ c\ OD\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ \{l\}\ Response\ action\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ c1\ ;;\ c2\ \{post\}\} \\
coms, p, aft \vdash \{\{pre\}\ Choose\ c1\ c2\ \{post\}\}
\end{array}$$

We tweak *fragments* by omitting *Responses*, yielding fewer obligations

**fun**

$$\begin{array}{l}
vcg\text{-}fragments' :: ('answer, 'location, 'question, 'state)\ com \\
\Rightarrow 'location\ set
\end{array}$$

$\Rightarrow ( ('answer, 'location, 'question, 'state) com$   
 $\times ('answer, 'location, 'question, 'state) loc-comp ) set$

**where**

$vcg\_fragments' (\{l\} Response\ action) aft = \{\}$   
 $| vcg\_fragments' (\{l\} IF\ b\ THEN\ c\ FI) aft$   
 $= vcg\_fragments' c aft$   
 $\cup \{ (\{l\} IF\ b\ THEN\ c' FI, lcond (atC\ c) aft\ b) | c'. True \}$   
 $| vcg\_fragments' (\{l\} IF\ b\ THEN\ c1\ ELSE\ c2\ FI) aft$   
 $= vcg\_fragments' c2 aft \cup vcg\_fragments' c1 aft$   
 $\cup \{ (\{l\} IF\ b\ THEN\ c1' ELSE\ c2' FI, lcond (atC\ c1) (atC\ c2) b) | c1' c2'. True \}$   
 $| vcg\_fragments' (LOOP\ DO\ c\ OD) aft = vcg\_fragments' c (atC\ c)$   
 $| vcg\_fragments' (\{l\} WHILE\ b\ DO\ c\ OD) aft$   
 $= vcg\_fragments' c \{l\} \cup \{ (\{l\} WHILE\ b\ DO\ c' OD, lcond (atC\ c) aft\ b) | c'. True \}$   
 $| vcg\_fragments' (c1 ;; c2) aft = vcg\_fragments' c2 aft \cup vcg\_fragments' c1 (atC\ c2)$   
 $| vcg\_fragments' (c1 \oplus c2) aft = vcg\_fragments' c1 aft \cup vcg\_fragments' c2 aft$   
 $| vcg\_fragments' c aft = \{(c, lconst\ aft)\}$

**abbreviation**

$vcg\_fragments :: ('answer, 'location, 'question, 'state) com$   
 $\Rightarrow ( ('answer, 'location, 'question, 'state) com$   
 $\times ('answer, 'location, 'question, 'state) loc-comp ) set$

**where**

$vcg\_fragments\ c \equiv vcg\_fragments' c \{\}$

**fun**  $isResponse :: ('answer, 'location, 'question, 'state) com \Rightarrow bool$  **where**

$isResponse (\{l\} Response\ action) \longleftrightarrow True$   
 $| isResponse - \longleftrightarrow False$

**lemma**  $fragments\text{-}vcg\_fragments'$ :

$\llbracket (c, aft) \in fragments\ c' aft'; \neg isResponse\ c \rrbracket \Longrightarrow (c, aft) \in vcg\_fragments' c' aft'$   
**by**  $(induct\ c' arbitrary: aft') auto$

**lemma**  $vcg\_fragments'\text{-}fragments$ :

$vcg\_fragments' c' aft' \subseteq fragments\ c' aft'$   
**by**  $(induct\ c' arbitrary: aft') (auto\ 10\ 0)$

**lemma**  $VCG\text{-}step$ :

**assumes**  $V: \bigwedge p. \forall (c, aft) \in vcg\_fragments\ (PGMs\ sys\ p). PGMs\ sys, p, aft \vdash \{pre\} c \{post\}$   
**assumes**  $S: system\text{-}step\ p\ sh' sh$   
**assumes**  $R: reachable\text{-}state\ sys\ sh$   
**assumes**  $P: pre\ sh$   
**shows**  $post\ sh'$

**using**  $S$

**proof**  $cases$

**case**  $LocalStep$  **with**  $P$  **show**  $?thesis$

**apply**  $-$

**apply**  $(erule\ decompose\text{-}small\text{-}step[OF - R])$

**apply**  $(frule\ fragments\text{-}basic\text{-}com)$

**apply**  $(erule\ basic\text{-}com.cases)$

**apply**  $(fastforce\ dest!:\ fragments\text{-}vcg\_fragments' V[rule\text{-}format])$

$elim: vcg\text{-}inv\ elim!:\ small\text{-}step\text{-}inv$

$simp: LST\text{-}def\ AT\text{-}def\ taken\text{-}def\ fun\text{-}eq\text{-}iff)+$

**done**

**next**

**case**  $CommunicationStep$  **with**  $P$  **show**  $?thesis$

**apply**  $-$

**apply**  $(erule\ decompose\text{-}small\text{-}step[OF - R])$

**apply**  $(erule\ decompose\text{-}small\text{-}step[OF - R])$

```

subgoal for  $c$   $cs$   $aft$   $c'$   $cs'$   $aft'$ 
apply (frule fragments-basic-com[where  $c'=c$ ])
apply (frule fragments-basic-com[where  $c'=c'$ ])
apply (elim basic-com.cases; clarsimp elim!: small-step-inv)
apply (drule fragments-vcg-fragments')
apply (fastforce dest!: V[rule-format]
      elim: vcg-inv elim!: small-step-inv
      simp: LST-def AT-def taken-def fun-eq-iff)+
done
done
qed

```

The user sees the conclusion of  $V$  for each element of *vcg-fragments*.

```

lemma VCG-step-inv-stable:
  assumes  $V: \bigwedge p. \forall (c, aft) \in \text{vcg-fragments} (PGMs \text{ sys } p). PGMs \text{ sys}, p, aft \vdash \{I\} c$ 
  assumes prerun sys  $\sigma$ 
  shows ( $[I] \leftrightarrow \circ[I]$ )  $\sigma$ 
apply (rule alwaysI)
apply clarsimp
apply (rule nextI)
apply clarsimp
using assms(2) unfolding prerun-def
apply clarsimp
apply (erule-tac i=i in alwaysE)
unfolding system-step-reflclp-def
apply clarsimp
apply (erule disjE; clarsimp)
using VCG-step[where pre=I and post=I] V assms(2) prerun-reachable-state
apply blast
done

```

```

lemma VCG:
  assumes  $I: \forall s. \text{initial-state sys } s \longrightarrow I (\emptyset GST = s, HST = \emptyset)$ 
  assumes  $V: \bigwedge p. \forall (c, aft) \in \text{vcg-fragments} (PGMs \text{ sys } p). PGMs \text{ sys}, p, aft \vdash \{I\} c$ 
  shows  $\text{sys} \models_{pre} I$ 
apply (rule prerun-valid-induct)
apply (clarsimp simp: prerun-def state-prop-def)
apply (metis (full-types) I old.unit.exhaust system-state.surjective)
using VCG-step-inv-stable[OF V] apply blast
done

```

**lemmas** *VCG-valid = valid-prerun-lift[OF VCG, of sys I] for sys I*

### 5.1.1 VCG rules

We can develop some (but not all) of the familiar Hoare rules; see [Lamport \(1980\)](#) and the `seL4/14.verified` lemma buckets for inspiration. We avoid many of the issues Lamport mentions as we only treat basic (atomic) commands.

```

context
  fixes coms :: ('answer, 'location, 'proc, 'question, 'state) programs
  fixes p :: 'proc
  fixes aft :: ('answer, 'location, 'question, 'state) loc-comp
begin

```

```

abbreviation
  valid-syn :: ('answer, 'location, 'proc, 'question, 'state) state-pred
     $\Rightarrow$  ('answer, 'location, 'question, 'state) com
     $\Rightarrow$  ('answer, 'location, 'proc, 'question, 'state) state-pred  $\Rightarrow$  bool where
  valid-syn  $P$   $c$   $Q \equiv \text{coms}, p, aft \vdash \{P\} c \{Q\}$ 

```

**notation** *valid-syn* ( $\{\cdot\}/ -/ \{\cdot\}$ )

**abbreviation**

*valid-inv-syn* :: ('answer, 'location, 'proc, 'question, 'state) state-pred  
 $\Rightarrow$  ('answer, 'location, 'question, 'state) com  $\Rightarrow$  bool **where**

*valid-inv-syn*  $P$   $c \equiv \{\{P\}\} c \{\{P\}\}$

**notation** *valid-inv-syn* ( $\{\cdot\}/ -$ )

**lemma** *vcg-True*:

$\{\{P\}\} c \{\{\langle True \rangle\}\}$

**by** (*cases*  $c$ ) (*fastforce elim!*: *vcg-inv intro: vcg.intros*) $+$

**lemma** *vcg-conj*:

$\llbracket \{\{I\}\} c \{\{Q\}\}; \{\{I\}\} c \{\{R\}\} \rrbracket \Longrightarrow \{\{I\}\} c \{\{Q \wedge R\}\}$

**by** (*cases*  $c$ ) (*fastforce elim!*: *vcg-inv intro: vcg.intros*) $+$

**lemma** *vcg-pre-imp*:

$\llbracket \wedge s. P s \Longrightarrow Q s; \{\{Q\}\} c \{\{R\}\} \rrbracket \Longrightarrow \{\{P\}\} c \{\{R\}\}$

**by** (*cases*  $c$ ) (*fastforce elim!*: *vcg-inv intro: vcg.intros*) $+$

**lemmas** *vcg-pre* = *vcg-pre-imp*[rotated]

**lemma** *vcg-post-imp*:

$\llbracket \wedge s. Q s \Longrightarrow R s; \{\{P\}\} c \{\{Q\}\} \rrbracket \Longrightarrow \{\{P\}\} c \{\{R\}\}$

**by** (*cases*  $c$ ) (*fastforce elim!*: *vcg-inv intro: vcg.intros*) $+$

**lemma** *vcg-prop*[*intro*]:

$\{\{\langle P \rangle\}\} c$

**by** (*cases*  $c$ ) (*fastforce intro: vcg.intros*) $+$

**lemma** *vcg-drop-imp*:

**assumes**  $\{\{P\}\} c \{\{Q\}\}$

**shows**  $\{\{P\}\} c \{\{R \longrightarrow Q\}\}$

**using** *assms*

**by** (*cases*  $c$ ) (*fastforce elim!*: *vcg-inv intro: vcg.intros*) $+$

**lemma** *vcg-conj-lift*:

**assumes**  $x: \{\{P\}\} c \{\{Q\}\}$

**assumes**  $y: \{\{P'\}\} c \{\{Q'\}\}$

**shows**  $\{\{P \wedge P'\}\} c \{\{Q \wedge Q'\}\}$

**apply** (*rule vcg-conj*)

**apply** (*rule vcg-pre*[*OF*  $x$ ], *simp*)

**apply** (*rule vcg-pre*[*OF*  $y$ ], *simp*)

**done**

**lemma** *vcg-disj-lift*:

**assumes**  $x: \{\{P\}\} c \{\{Q\}\}$

**assumes**  $y: \{\{P'\}\} c \{\{Q'\}\}$

**shows**  $\{\{P \vee P'\}\} c \{\{Q \vee Q'\}\}$

**using** *assms*

**by** (*cases*  $c$ ) (*fastforce elim!*: *vcg-inv intro: vcg.intros*) $+$

**lemma** *vcg-imp-lift*:

**assumes**  $\{\{P'\}\} c \{\{\neg P\}\}$

**assumes**  $\{\{Q'\}\} c \{\{Q\}\}$

**shows**  $\{\{P' \vee Q'\}\} c \{\{P \longrightarrow Q\}\}$

**by** (*simp only: imp-conv-disj vcg-disj-lift*[*OF* *assms*])



**lemma** *vcg-ex-lift*:  
**assumes**  $\bigwedge x. \{P\ x\} \ c \ \{Q\ x\}$   
**shows**  $\{\lambda s. \exists x. P\ x\ s\} \ c \ \{\lambda s. \exists x. Q\ x\ s\}$   
**using** *assms*  
**by** (*cases c*) (*fastforce elim!: vcg-inv intro: vcg.intros*)+

**lemma** *vcg-all-lift*:  
**assumes**  $\bigwedge x. \{P\ x\} \ c \ \{Q\ x\}$   
**shows**  $\{\lambda s. \forall x. P\ x\ s\} \ c \ \{\lambda s. \forall x. Q\ x\ s\}$   
**using** *assms*  
**by** (*cases c*) (*fastforce elim!: vcg-inv intro: vcg.intros*)+

**lemma** *vcg-name-pre-state*:  
**assumes**  $\bigwedge s. P\ s \implies \{(\ = \) s\} \ c \ \{Q\}$   
**shows**  $\{P\} \ c \ \{Q\}$   
**using** *assms*  
**by** (*cases c*) (*fastforce elim!: vcg-inv intro: vcg.intros*)+

**lemma** *vcg-lift-comp*:  
**assumes**  $f: \bigwedge P. \{\lambda s. P\ (f\ s :: 'a :: type)\} \ c$   
**assumes**  $P: \bigwedge x. \{Q\ x\} \ c \ \{P\ x\}$   
**shows**  $\{\lambda s. Q\ (f\ s)\ s\} \ c \ \{\lambda s. P\ (f\ s)\ s\}$   
**apply** (*rule vcg-name-pre-state*)  
**apply** (*rename-tac s*)  
**apply** (*rule vcg-pre*)  
**apply** (*rule vcg-post-imp[rotated]*)  
**apply** (*rule vcg-conj-lift*)  
**apply** (*rule-tac x=f s in P*)  
**apply** (*rule-tac P= $\lambda f s. fs = f s$  in f*)  
**apply** *simp*  
**apply** *simp*  
**done**

### 5.1.2 Cheap non-interference rules

These rules magically construct VCG lifting rules from the easier to prove *eq-imp* facts. We don't actually use these in the GC, but we do derive *fun-upd* equations using the same mechanism. Thanks to Thomas Sewell for the requisite syntax magic.

As these *eq-imp* facts do not usefully compose, we make the definition asymmetric (i.e., *g* does not get a bundle of parameters).

Note that these are effectively parametricity rules.

**definition** *eq-imp* ::  $('a \Rightarrow 'b \Rightarrow 'c) \Rightarrow ('b \Rightarrow 'e) \Rightarrow \text{bool}$  **where**  
 $\text{eq-imp } f\ g \equiv (\forall s\ s'. (\forall x. f\ x\ s = f\ x\ s') \longrightarrow (g\ s = g\ s'))$

**lemma** *eq-impD*:  
 $\llbracket \text{eq-imp } f\ g; \forall x. f\ x\ s = f\ x\ s' \rrbracket \implies g\ s = g\ s'$   
**by** (*simp add: eq-imp-def*)

**lemma** *eq-imp-vcg*:  
**assumes**  $g: \text{eq-imp } f\ g$   
**assumes**  $f: \forall x\ P. \{P \circ (f\ x)\} \ c$   
**shows**  $\{P \circ g\} \ c$   
**apply** (*rule vcg-name-pre-state*)  
**apply** (*rename-tac s*)  
**apply** (*rule vcg-pre*)  
**apply** (*rule vcg-post-imp[rotated]*)  
**apply** (*rule vcg-all-lift[where 'a='a]*)

```

apply (rule-tac  $x=x$  and  $P=\lambda fs. fs = f x s$  in  $f[rule-format]$ )
apply simp
apply (frule eq-impD[where  $f=f, OF g$ ])
apply simp
apply simp
done

```

```

lemma eq-imp-vcg-LST:
  assumes  $g: eq-imp f g$ 
  assumes  $f: \forall x P. \{P \circ (f x) \circ LST\} c$ 
  shows  $\{P \circ g \circ LST\} c$ 
apply (rule vcg-name-pre-state)
apply (rename-tac  $s$ )
apply (rule vcg-pre)
apply (rule vcg-post-imp[rotated])
  apply (rule vcg-all-lift[where  $'a='a'$ ])
  apply (rule-tac  $x=x$  and  $P=\lambda fs. fs = f x s \downarrow$  in  $f[rule-format]$ )
apply simp
apply (frule eq-impD[where  $f=f, OF g$ ])
apply simp
apply simp
done

```

```

lemma eq-imp-fun-upd:
  assumes  $g: eq-imp f g$ 
  assumes  $f: \forall x. f x (s(fld := val)) = f x s$ 
  shows  $g (s(fld := val)) = g s$ 
apply (rule eq-impD[ $OF g$ ])
apply (rule  $f$ )
done

```

```

lemma curry-forall-eq:
   $(\forall f. P f) = (\forall f. P (case-prod f))$ 
by (metis case-prod-curry)

```

```

lemma pres-tuple-vcg:
   $(\forall P. \{P \circ (\lambda s. (f s, g s))\} c)$ 
   $\longleftrightarrow ((\forall P. \{P \circ f\} c) \wedge (\forall P. \{P \circ g\} c))$ 
apply (simp add: curry-forall-eq o-def)
apply safe
  apply fast
  apply fast
apply (rename-tac  $P$ )
apply (rule-tac  $f=f$  and  $P=\lambda fs s. P fs (g s)$  in  $vcg-lift-comp; simp$ )
done

```

```

lemma pres-tuple-vcg-LST:
   $(\forall P. \{P \circ (\lambda s. (f s, g s)) \circ LST\} c)$ 
   $\longleftrightarrow ((\forall P. \{P \circ f \circ LST\} c) \wedge (\forall P. \{P \circ g \circ LST\} c))$ 
apply (simp add: curry-forall-eq o-def)
apply safe
  apply fast
  apply fast
apply (rename-tac  $P$ )
apply (rule-tac  $f=\lambda s. f s \downarrow$  and  $P=\lambda fs s. P fs (g s)$  for  $g$  in  $vcg-lift-comp; simp$ )
done

```

```

no-notation valid-syn ( $\{-\}/ -/ \{-\}$ )

```

end

## 6 One locale per process

A sketch of what we're doing in *ConcurrentGC*, for quicker testing.  
FIXME write some lemmas that further exercise the generated thms.

locale *P1*  
begin

**definition** *com* :: (unit, string, unit, nat) com **where**  
  *com* = {"A"} WHILE ((<) 0) DO {"B"} [λs. s - 1] OD

**intern-com** *com-def*  
**print-theorems**

**locset-definition** *loop* = {*B*}  
**print-theorems**  
**thm** *locset-cache*

**definition** *assertion* = atS False *loop*

end

**thm** *locset-cache*

locale *P2*  
begin

**thm** *locset-cache*

**definition** *com* :: (unit, string, unit, nat) com **where**  
  *com* = {"C"} WHILE ((<) 0) DO {"A"} [Suc] OD

**intern-com** *com-def*  
**locset-definition** *loop* = {*A*}  
**print-theorems**

end

**thm** *locset-cache*

**primrec** *coms* :: bool ⇒ (unit, string, unit, nat) com **where**  
  *coms* False = *P1.com*  
  | *coms* True = *P2.com*

## 7 Example: a one-place buffer

To demonstrate the CIMP reasoning infrastructure, we treat the trivial one-place buffer example of [Lamport and Schneider \(1984, §3.3\)](#). Note that the semantics for our language is different to [Lamport and Schneider's](#), who treated a historical variant of CSP (i.e., not the one in [Hoare \(1985\)](#)).

We introduce some syntax for fixed-topology (static channel-based) scenarios.

**abbreviation**

*rcv-syn* :: 'location ⇒ 'channel ⇒ ('val ⇒ 'state ⇒ 'state)

$$\Rightarrow (\text{unit}, \text{'location}, \text{'channel} \times \text{'val}, \text{'state}) \text{ com } (\{\!|\cdot|\!\}/ \dashv\!- [0,0,81] 81)$$

where

$$\{\!|l|\!\} \text{ ch} \triangleright f \equiv \{\!|l|\!\} \text{ Response } (\lambda q \text{ s. if fst } q = \text{ch then } \{(f (\text{snd } q) \text{ s}, ())\} \text{ else } \{\})$$

abbreviation

$$\begin{aligned} \text{snd-syn} &:: \text{'location} \Rightarrow \text{'channel} \Rightarrow (\text{'state} \Rightarrow \text{'val}) \\ &\Rightarrow (\text{unit}, \text{'location}, \text{'channel} \times \text{'val}, \text{'state}) \text{ com } (\{\!|\cdot|\!\}/ \dashv\!- [0,0,81] 81) \end{aligned}$$

where

$$\{\!|l|\!\} \text{ ch} \triangleleft f \equiv \{\!|l|\!\} \text{ Request } (\lambda s. (\text{ch}, f \text{ s})) (\lambda \text{ans } s. \{s\})$$

These definitions largely follow [Lamport and Schneider \(1984\)](#). We have three processes communicating over two channels. We enumerate program locations.

**datatype**  $\text{ex-chname} = \xi 12 \mid \xi 23$

**type-synonym**  $\text{ex-val} = \text{nat}$

**type-synonym**  $\text{ex-ch} = \text{ex-chname} \times \text{ex-val}$

**datatype**  $\text{ex-loc} = r12 \mid r23 \mid s23 \mid s12$

**datatype**  $\text{ex-proc} = p1 \mid p2 \mid p3$

**type-synonym**  $\text{ex-pgm} = (\text{unit}, \text{ex-loc}, \text{ex-ch}, \text{ex-val}) \text{ com}$

**type-synonym**  $\text{ex-pred} = (\text{unit}, \text{ex-loc}, \text{ex-proc}, \text{ex-ch}, \text{ex-val}) \text{ state-pred}$

**type-synonym**  $\text{ex-state} = (\text{unit}, \text{ex-loc}, \text{ex-proc}, \text{ex-ch}, \text{ex-val}) \text{ system-state}$

**type-synonym**  $\text{ex-sys} = (\text{unit}, \text{ex-loc}, \text{ex-proc}, \text{ex-ch}, \text{ex-val}) \text{ system}$

**type-synonym**  $\text{ex-history} = (\text{ex-ch} \times \text{unit}) \text{ list}$

We further specialise these for our particular example.

**primrec**

$$\text{ex-coms} :: \text{ex-proc} \Rightarrow \text{ex-pgm}$$

where

$$\text{ex-coms } p1 = \{\!|s12|\!\} \xi 12 \triangleleft \text{id}$$

$$\mid \text{ex-coms } p2 = \text{LOOP DO } \{\!|r12|\!\} \xi 12 \triangleright (\lambda v \text{ . } v) ;; \{\!|s23|\!\} \xi 23 \triangleleft \text{id OD}$$

$$\mid \text{ex-coms } p3 = \{\!|r23|\!\} \xi 23 \triangleright (\lambda v \text{ . } v)$$

Each process starts with an arbitrary initial local state.

**abbreviation**  $\text{ex-init} :: (\text{ex-proc} \Rightarrow \text{ex-val}) \Rightarrow \text{bool}$  **where**

$$\text{ex-init} \equiv \langle \text{True} \rangle$$

**abbreviation**  $\text{sys} :: \text{ex-sys}$  **where**

$$\text{sys} \equiv (\text{PGMs} = \text{ex-coms}, \text{INIT} = \text{ex-init}, \text{FAIR} = \langle \text{True} \rangle)$$

The following adapts Kai Engelhardt's, from his notes titled *Proving an Asynchronous Message Passing Program Correct*, 2011. The history variable tracks the causality of the system, which I feel is missing in Lamport's treatment. We tack on Lamport's invariant so we can establish *Etern-pred*.

**abbreviation**

$$\text{filter-on-channel} :: \text{ex-chname} \Rightarrow \text{ex-state} \Rightarrow \text{ex-val list } (\_ \text{ } [100] 101)$$

where

$$\_ \text{ ch} \equiv \text{map } (\text{snd} \circ \text{fst}) \circ \text{filter } ((=) \text{ ch} \circ \text{fst} \circ \text{fst}) \circ \text{HST}$$

**definition**  $\text{IL} :: \text{ex-pred}$  **where**

$$\begin{aligned} \text{IL} &= \text{pred-conjoin } [ \\ &\quad \text{at } p1 \text{ } s12 \longrightarrow \text{LIST-NULL } \mid \xi 12 \\ &\quad , \text{terminated } p1 \longrightarrow \_ \mid \xi 12 = (\lambda s. [s \downarrow p1]) \\ &\quad , \text{at } p2 \text{ } r12 \longrightarrow \_ \mid \xi 12 = \_ \mid \xi 23 \\ &\quad , \text{at } p2 \text{ } s23 \longrightarrow \_ \mid \xi 12 = \_ \mid \xi 23 \text{ @ } (\lambda s. [s \downarrow p2]) \wedge (\lambda s. s \downarrow p1 = s \downarrow p2) \\ &\quad , \text{at } p3 \text{ } r23 \longrightarrow \text{LIST-NULL } \mid \xi 23 \\ &\quad , \text{terminated } p3 \longrightarrow \_ \mid \xi 23 = (\lambda s. [s \downarrow p2]) \wedge (\lambda s. s \downarrow p1 = s \downarrow p3) \\ &] \end{aligned}$$

If  $p3$  terminates, then it has  $p1$ 's value. This is stronger than [Lamport and Schneider's](#) as we don't ask that the first process has also terminated.

**definition** *Etern-pred* :: *ex-pred* **where**

*Etern-pred* = (*terminated* *p3*  $\longrightarrow$  ( $\lambda s. s \downarrow p1 = s \downarrow p3$ ))

Proofs from here down.

**lemma** *correct-system*:

**assumes** *IL sh*

**shows** *Etern-pred sh*

**using** *assms unfolding Etern-pred-def IL-def* **by** *simp*

**lemma** *IL-p1*: *ex-coms*, *p1*, *lconst*  $\{\}$   $\vdash$   $\{\{IL\}\}$   $\{\{s12\}\}$   $\xi12 \triangleleft (\lambda s. s)$

**apply** (*rule vcg.intros*)

**apply** (*rename-tac p'*)

**apply** (*case-tac p'*; *clarsimp simp: IL-def atLs-def*)

**done**

**lemma** *IL-p2*: *ex-coms*, *p2*, *lconst*  $\{r12\}$   $\vdash$   $\{\{IL\}\}$   $\{\{s23\}\}$   $\xi23 \triangleleft (\lambda s. s)$

**apply** (*rule vcg.intros*)

**apply** (*rename-tac p'*)

**apply** (*case-tac p'*; *clarsimp simp: IL-def*)

**done**

**lemma** *IL*: *sys*  $\models_{pre}$  *IL*

**apply** (*rule VCG*)

**apply** (*clarsimp simp: IL-def atLs-def dest!: initial-stateD*)

**apply** (*rename-tac p*)

**apply** (*case-tac p*; *clarsimp simp: IL-p1 IL-p2*)

**done**

**lemma** *IL-valid*: *sys*  $\models \square[IL]$

**by** (*rule valid-prerun-lift[OF IL]*)

## 8 Example: an unbounded buffer

This is more literally Kai Engelhardt's example from his notes titled *Proving an Asynchronous Message Passing Program Correct*, 2011.

**datatype** *ex-chname* =  $\xi12 \mid \xi23$

**type-synonym** *ex-val* = *nat*

**type-synonym** *ex-ls* = *ex-val* *list*

**type-synonym** *ex-ch* = *ex-chname*  $\times$  *ex-val*

**datatype** *ex-loc* = *c1*  $\mid$  *r12*  $\mid$  *r23*  $\mid$  *s23*  $\mid$  *s12*

**datatype** *ex-proc* = *p1*  $\mid$  *p2*  $\mid$  *p3*

**type-synonym** *ex-pgm* = (*unit*, *ex-loc*, *ex-ch*, *ex-ls*) *com*

**type-synonym** *ex-pred* = (*unit*, *ex-loc*, *ex-proc*, *ex-ch*, *ex-ls*) *state-pred*

**type-synonym** *ex-state* = (*unit*, *ex-loc*, *ex-proc*, *ex-ch*, *ex-ls*) *system-state*

**type-synonym** *ex-sys* = (*unit*, *ex-loc*, *ex-proc*, *ex-ch*, *ex-ls*) *system*

**type-synonym** *ex-history* = (*ex-ch*  $\times$  *unit*) *list*

The local state for the producer process contains all values produced; consider that ghost state.

**abbreviation** (*input*) *snoc* :: '*a*  $\Rightarrow$  '*a* *list*  $\Rightarrow$  '*a* *list* **where** *snoc* *x xs*  $\equiv$  *xs* @ [*x*]

**primrec** *ex-coms* :: *ex-proc*  $\Rightarrow$  *ex-pgm* **where**

*ex-coms* *p1* = *LOOP DO*  $\{\{c1\}\}$  *LocalOp* ( $\lambda xs. \{\{snoc\} x xs \mid x. True\}$ ) ;;  $\{\{s12\}\}$   $\xi12 \triangleleft (last, id)$  *OD*  
 $\mid$  *ex-coms* *p2* = *LOOP DO*  $\{\{r12\}\}$   $\xi12 \triangleright snoc$   
 $\oplus \{\{c1\}\}$  *IF* ( $\lambda s. length\ s > 0$ ) *THEN*  $\{\{s23\}\}$   $\xi12 \triangleleft (hd, tl)$  *FI*  
*OD*

|  $ex\text{-}coms\ p3 = LOOP\ DO\ \{\{r23\}\}\ \xi23 \triangleright snoc\ OD$

**abbreviation**  $ex\text{-}init :: (ex\text{-}proc \Rightarrow ex\text{-}ls) \Rightarrow bool$  **where**  
 $ex\text{-}init\ s \equiv \forall p. s\ p = []$

**abbreviation**  $sys :: ex\text{-}sys$  **where**  
 $sys \equiv (\{PGMs = ex\text{-}coms, INIT = ex\text{-}init, FAIR = \langle True \rangle\})$

**abbreviation**  
 $filter\text{-}on\text{-}channel :: ex\text{-}chname \Rightarrow ex\text{-}state \Rightarrow ex\text{-}val\ list\ (\_ - [100] 101)$   
**where**  
 $\_ ch \equiv map\ (snd \circ fst) \circ filter\ ((=)\ ch \circ fst \circ fst) \circ HST$

**definition**  $I\text{-}pred :: ex\text{-}pred$  **where**  
 $I\text{-}pred = pred\text{-}conjoin\ [$   
 $\quad at\ p1\ c1 \longrightarrow \_ \xi12 = (\lambda s. s \downarrow p1)$   
 $\quad ,\ at\ p1\ s12 \longrightarrow (\lambda s. length\ (s \downarrow p1) > 0 \wedge butlast\ (s \downarrow p1) = (\_ \xi12)\ s)$   
 $\quad ,\ \_ \xi12 \leq (\lambda s. s \downarrow p1)$   
 $\quad ,\ \_ \xi12 = \_ \xi23\ @\ (\lambda s. s \downarrow p2)$   
 $\quad ,\ at\ p2\ s23 \longrightarrow (\lambda s. length\ (s \downarrow p2) > 0)$   
 $\quad ,\ (\lambda s. s \downarrow p3) = \_ \xi23$   
 $\quad ]$

The local state of  $p3$  is some prefix of the local state of  $p1$ .

**definition**  $Etern\text{-}pred :: ex\text{-}pred$  **where**  
 $Etern\text{-}pred \equiv \lambda s. s \downarrow p3 \leq s \downarrow p1$

**lemma**  $correct\text{-}system$ :  
**assumes**  $I\text{-}pred\ s$   
**shows**  $Etern\text{-}pred\ s$   
**using**  $assms\ unfolding\ Etern\text{-}pred\text{-}def\ I\text{-}pred\text{-}def\ less\text{-}eq\text{-}list\text{-}def\ prefix\text{-}def$  **by**  $clarsimp$

**lemma**  $p1\text{-}c1[simplified, intro]$ :  
 $ex\text{-}coms, p1, lconst\ \{c1\} \vdash \{\{I\text{-}pred\}\}\ \{\{c1\}\}\ LocalOp\ (\lambda xs. \{ snoc\ x\ xs\ |x.\ True\ \})$   
**apply**  $(rule\ vcg.intros)$   
**apply**  $(clarsimp\ simp: I\text{-}pred\text{-}def\ atS\text{-}def)$   
**done**

**lemma**  $p1\text{-}s12[simplified, intro]$ :  
 $ex\text{-}coms, p1, lconst\ \{c1\} \vdash \{\{I\text{-}pred\}\}\ \{\{s12\}\}\ \xi12 \triangleleft (last, id)$   
**apply**  $(rule\ vcg.intros)$   
**apply**  $(rename\text{-}tac\ p')$   
**apply**  $(case\text{-}tac\ p'; clarsimp)$   
**apply**  $(clarsimp\ simp: I\text{-}pred\text{-}def\ atS\text{-}def)$   
**apply**  $(metis\ Prefix\text{-}Order.prefix\text{-}snoc\ append.assoc\ append\text{-}butlast\text{-}last\text{-}id)$   
**done**

**lemma**  $p2\text{-}s23[simplified, intro]$ :  
 $ex\text{-}coms, p2, lconst\ \{c1, r12\} \vdash \{\{I\text{-}pred\}\}\ \{\{s23\}\}\ \xi12 \triangleleft (hd, tl)$   
**apply**  $(rule\ vcg.intros)$   
**apply**  $(rename\text{-}tac\ p')$   
**apply**  $(case\text{-}tac\ p'; clarsimp)$   
**done**

**lemma**  $p2\text{-}pi4[intro]$ :  
 $ex\text{-}coms, p2, lcond\ \{s23\}\ \{c1, r12\} (\lambda s. s \neq []) \vdash \{\{I\text{-}pred\}\}\ \{\{c1\}\}\ IF\ (\lambda s. s \neq [])\ THEN\ c'\ FI$   
**apply**  $(rule\ vcg.intros)$   
**apply**  $(clarsimp\ simp: I\text{-}pred\text{-}def\ atS\text{-}def\ split: lcond\text{-}splits)$

done

```
lemma I: sys  $\models_{pre}$  I-pred
apply (rule VCG)
  apply (clarsimp dest!: initial-stateD simp: I-pred-def atS-def)
apply (rename-tac p)
apply (case-tac p; auto)
done
```

```
lemma I-valid: sys  $\models \square[I-pred]$ 
by (rule valid-prerun-lift[OF I])
```

## 9 Concluding remarks

Previously Nipkow and Prensa Nieto (1999); Prensa Nieto (2002, 2003)<sup>3</sup> have developed the classical Owicki/Gries and Rely-Guarantee paradigms for the verification of shared-variable concurrent programs in Isabelle/HOL. These have been used to show the correctness of a garbage collector (Prensa Nieto and Esparza 2000).

We instead use synchronous message passing, which is significantly less explored. de Boer, de Roever, and Hannemann (1999); ? provide compositional systems for *terminating* systems. We have instead adopted Lamport's paradigm of a single global invariant and local proof obligations as the systems we have in mind are tightly coupled and it is not obvious that the proofs would be easier on a decomposed system; see ?, §1.6.6 for a concurring opinion. Unlike the generic sequential program verification framework Simpl (Schirmer 2004), we do not support function calls, or a sophisticated account of state spaces. Moreover we do no meta-theory beyond showing the simple VCG is sound (§5.1).

## References

- B. Alpern and F. B. Schneider. Defining liveness. *Information Processing Letters*, 21(4):181–185, 1985. doi: 10.1016/0020-0190(85)90056-0.
- K. R. Apt, N. Francez, and W. P. de Roever. A proof system for communicating sequential processes. *ACM Transactions on Programming Languages and Systems*, 2(3):359–385, 1980. doi: 10.1145/357103.357110.
- S. Berghofer. A solution to the PoplMark challenge using de Bruijn indices in Isabelle/HOL. *J. Autom. Reasoning*, 49(3):303–326, 2012.
- K. M. Chandy and J. Misra. *Parallel program design - a foundation*. Addison-Wesley, 1989. ISBN 978-0-201-05866-6.
- E. Y. Chang, Z. Manna, and A. Pnueli. Characterization of temporal property classes. In *ICALP'1992*, volume 623 of *LNCS*, pages 474–486. Springer, 1992. doi: 10.1007/3-540-55719-9\\_97.
- P. Cousot and R. Cousot. Semantic analysis of Communicating Sequential Processes (shortened version). In *ICALP*, volume 85 of *LNCS*, pages 119–133. Springer, 1980.
- P. Cousot and R. Cousot. A language independent proof of the soundness and completeness of generalized Hoare logic. *Information and Computation*, 80(2):165–191, February 1989.
- F. S. de Boer, W. P. de Roever, and U. Hannemann. The semantic foundations of a compositional proof method for synchronously communicating processes. In M. Kutylowski, L. Pacholski, and T. Wierzbicki, editors, *MFCS*, volume 1672 of *LNCS*, pages 343–353. Springer, 1999.
- W. P. de Roever, F. S. de Boer, U. Hannemann, J. Hooman, Y. Lakhnech, M. Poel, and J. Zwiers. *Concurrency Verification: Introduction to Compositional and Noncompositional Methods*, volume 54 of *Cambridge Tracts in Theoretical Computer Science*. Cambridge University Press, 2001.

---

<sup>3</sup>The theories are in \$ISABELLE/src/HOL/Hoare\_Parallel.

- M. Felleisen and R. Hieb. The revised report on the syntactic theories of sequential control and state. *Theoretical Computer Science*, 103(2):235–271, 1992. doi: 10.1016/0304-3975(92)90014-7.
- G. Grov and S. Merz. A definitional encoding of tla\* in isabelle/hol. *Archive of Formal Proofs*, November 2011. ISSN 2150-914x. <http://isa-afp.org/entries/TLA>, Formal proof development.
- C.A.R. Hoare. *Communicating Sequential Processes*. International Series In Computer Science. Prentice-Hall, 1985. URL <http://www.usingscp.com/>.
- P. B. Jackson. Verifying a garbage collection algorithm. In *TPHOLs*, volume 1479 of *LNCS*, pages 225–244. Springer, 1998. doi: 10.1007/BFb0055139.
- E. Kindler. Safety and liveness properties: A survey. *Bulletin of the European Association for Theoretical Computer Science*, 53(30):268–272, 6 1994. URL <http://citeseerx.ist.psu.edu/viewdoc/download?doi=10.1.1.43.8206&rep=rep1&type=pdf>.
- L. Lamport. The “Hoare Logic” of concurrent programs. *Acta Informatica*, 14:21–37, 1980.
- L. Lamport. *Specifying Systems, The TLA+ Language and Tools for Hardware and Software Engineers*. Addison-Wesley, 2002.
- L. Lamport and F. B. Schneider. The “Hoare Logic” of CSP, and all that. *ACM Transactions on Programming Languages and Systems*, 6(2):281–296, 1984.
- G. Levin and D. Gries. A proof technique for communicating sequential processes. *Acta Inf.*, 15:281–302, 1981.
- Z. Manna and A. Pnueli. The anchored version of the temporal framework. In J. W. de Bakker, W. P. de Roever, and G. Rozenberg, editors, *Linear Time, Branching Time and Partial Order in Logics and Models for Concurrency, School/Workshop, Noordwijkerhout, The Netherlands, May 30 - June 3, 1988, Proceedings*, volume 354 of *LNCS*, pages 201–284. Springer, 1988. doi: 10.1007/BFb0013024.
- Z. Manna and A. Pnueli. Tools and rules for the practicing verifier. In R. F. Rashid, editor, *CMU Computer Science: A 25th Anniversary Commemorative*, pages 121–156. ACM Press and Addison-Wesley, 1991. Also Technical Report STAN-CS-90-1321.
- Z. Manna and A. Pnueli. *Temporal verification of reactive systems - Safety*. Springer, 1995.
- Z. Manna and A. Pnueli. Temporal verification of reactive systems: Response. In *Time for Verification, Essays in Memory of Amir Pnueli*, volume 6200 of *LNCS*, pages 279–361. Springer, 2010. doi: 10.1007/978-3-642-13754-9\\_13.
- R. Milner. *A Calculus of Communicating Systems*. Springer, 1980.
- R. Milner. *Communication and Concurrency*. Prentice-Hall, Englewood Cliffs, NJ, 1989.
- T. Nipkow and G. Klein. *Concrete Semantics: A Proof Assistant Approach*. Springer, 2014. URL <http://www.in.tum.de/~nipkow/Concrete-Semantics/>.
- T. Nipkow and L. Prensa Nieto. Owicki/Gries in Isabelle/HOL. In J.-P. Finance, editor, *FASE*, volume 1577 of *LNCS*, pages 188–203. Springer, 1999.
- S. S. Owicki and L. Lamport. Proving liveness properties of concurrent programs. *ACM Transactions on Programming Languages and Systems*, 4(3):455–495, 1982. doi: 10.1145/357172.357178.
- A. M. Pitts. Operational semantics and program equivalence. In G. Barthe, P. Dybjer, and J. Saraiva, editors, *Applied Semantics, Advanced Lectures*, volume 2395 of *LNCS*, pages 378–412. Springer, 2002. International Summer School, APPSEM 2000, Caminha, Portugal, September 9–15, 2000.
- G. D. Plotkin. The origins of structural operational semantics. *Journal of Logic and Algebraic Programming*, 60-61:3–15, 2004.
- L. Prensa Nieto. *Verification of Parallel Programs with the Owicki-Gries and Rely-Guarantee Methods in Isabelle/HOL*. PhD thesis, Technische Universität München, 2002.
- L. Prensa Nieto. The Rely-Guarantee method in Isabelle/HOL. In P. Degano, editor, *ESOP’2003*, volume 2618 of *LNCS*, pages 348–362. Springer, 2003.



- L. Prensa Nieto and J. Esparza. Verifying single and multi-mutator garbage collectors with Owicki/Gries in Isabelle/HOL. In M. Nielsen and B. Rovan, editors, *Mathematical Foundations of Computer Science (MFCS 2000)*, volume 1893 of *LNCS*, pages 619–628. Springer, 2000.
- T. Ridge. Verifying distributed systems: the operational approach. In *POPL'2009*, pages 429–440. ACM, 2009. doi: 10.1145/1480881.1480934.
- N. Schirmer. A verification environment for sequential imperative programs in isabelle/hol. In F. Baader and A. Voronkov, editors, *LPAR*, volume 3452 of *LNCS*, pages 398–414. Springer, 2004.
- N. Schirmer and M. Wenzel. State spaces - the locale way. *Electr. Notes Theor. Comput. Sci.*, 254:161–179, 2009.
- F. B. Schneider. Decomposing properties into safety and liveness using predicate logic. Technical Report 87-874, Department of Computer Science, Cornell University, October 1987.
- A. P. Sistla. Safety, liveness and fairness in temporal logic. *Formal Aspects of Computing*, 6(5):495–512, 1994. doi: 10.1007/BF01211865.
- R. J. van Glabbeek and P. Höfner. Progress, justness and fairness. *ACM Computing Surveys*, 2019. URL <http://arxiv.org/abs/1810.07414v1>.
- G. Winskel. *The Formal Semantics of Programming Languages*. MIT Press, Cambridge, MA, 1993.